

"Aladino e le Perle di Saggezza" by Leonardo Boselli

```
[
*** Versioni ***

Versione 1      semi-definitiva
Versione 2      l'asino non fugge più dalla valle delle tombe,
                nè si libera dalla fune quando compare l'orda;
                fissata la carogna
Versione 3      fissati tutti i contenitori mobili (il tino, l'arnia, il nido)
Versione 4      sostituito "Trinity Inventory" con "Written Inventory"
                aggiunta "Keyword Interface" alla "Blue Lacuna"
                modificata con scenery e dialoghi
                apriti sesamo efficace solo davanti alla grotta
                necessario risolvere il cubo prima di darlo al capo
                saggio risponde solo se visibile
                immagini e suoni
Versione 5      descrizione acquaiolo
                plurale orda
                collegamenti ipertestuali
                azioni dopo esamina
Versione 6      libretto tradotto
                sistemato vicolo cieco senza cammello nel deserto di rocce
                corretto intreccia matassa e altri verbi
                (aggiunto salva, carica e "x me" in menù Hyperlink Interface IT
                per evitare il problema di alcuni interpreti che vanno in crash
                quando si scrive con la tastiera dopo aver usato gli hyperlink)

*** Note per la compilazione ***
```

"Aladino e le Perle di Saggezza" è disponibile in tre versioni: "Pure", "Deluxe" e "Hyper". Dato che non è disponibile in Inform7 un preprocessore (almeno che io sappia), ho utilizzato un sistema di commenti che includa o escluda alcune parti di codice. Per questo motivo nel sorgente compaiono varie parole chiave in apertura e chiusura delle parti di codice da includere/escludere. Ad esempio %\$ e %\$, KEYW% e %KEYW, HYPER% e %HYPER.

- Versione "Pure"

si tratta di una versione senza immagini e suoni che può essere compilata con o senza le parole chiave evidenziate alla Blue Lacuna (un metodo per enfatizzare con colori o stili diversi le parole che corrispondono a oggetti da esaminare, direzioni da percorrere e argomenti di discussione).

- Versione "Deluxe"

si tratta di una versione con immagini, suoni e le parole chiave evidenziate alla Blue Lacuna.

- Versione "Hyper"

si tratta di una versione con immagini, suoni e collegamenti ipertestuali cliccabili al posto delle parole chiave evidenziate alla Blue Lacuna.

Il sorgente qui presentato consente di compilare la versione "Hyper". Per le altre sono necessarie le seguenti modifiche:

- * per eliminare le immagini ed i suoni occorre trasformare i testi (\$%) e (%\$) in (%\$ e %\$) con le parentesi quadre al posto di quelle tonde;
- * per eliminare i link ipertestuali occorre trasformare i testi (HYPER%) e (%HYPER) in (HYPER% e %HYPER), sempre con le parentesi quadre al posto di quelle tonde;
- * per aggiungere le parole chiave alla "Blue Lacuna" occorre trasformare i testi (KEYW% e %KEYW) in (KEYW%) e (%KEYW), ancora con le parentesi quadre al posto di quelle tonde.

Part I - Impostazioni iniziali**Chapter 1 - Titolo**

The story headline is "Una storia da Mille e Una Avventura".

The story genre is "surreale".

The release number is 6.

The story description is "Aladino, in viaggio nel deserto al seguito di una carovana, viene assalito dai predoni, derubato dei suoi averi e lasciato agonizzare ai margini di un'oasi. Nel tentativo di ritrovare la strada di casa, conoscerà alcuni personaggi che l'aiuteranno a rimettersi in sesto. In particolare incontrerà un vecchio saggio che, con accorti suggerimenti, gli indicherà il cammino del ritorno."

The story creation year is 2009.

[\$%] Release along with a "Italian" website, cover art and source text [%\$].

Chapter 2 - Includes

Include Italian by Leonardo Boselli.

[\$%] Include Title Page IT by Leonardo Boselli. [%\$]

Include Creative Commons Public License IT by Leonardo Boselli.

```

Include Basic Help Menu IT by Leonardo Boselli.
Include Written Inventory IT by Leonardo Boselli.
Include Remembering IT by Leonardo Boselli.
Include Underside IT by Leonardo Boselli.
Include Rideable Vehicles IT by Leonardo Boselli.
Include Conversation Package IT by Leonardo Boselli.
[KEYW% Include Keyword Interface IT by Leonardo Boselli. %KEYW]
[HYPER%] Include Hyperlink Interface IT by Leonardo Boselli [%HYPER]

```

Chapter 3 - Enfattizzazione

```

First when play begins:
[HYPER%]
    now topic hyperlink highlighting is true;
    now Menu Hyperlink shown is true.
[%HYPER]
[KEYW%]
    now topic keyword highlighting is true;
    now parser highlighting is true;
%KEYW]

```

Chapter 4 - Help

Table of Basic Help Options (continued)

title	subtable	description
[HYPER%]		
"[bold type]Collegamenti ipertestuali[fixed letter spacing]"	--	"[show-HI-message]"
[HYPER%]		
[KEYW%]		
"[bold type]Parole chiave enfattizzate[fixed letter spacing]"	--	"[show-KI-message]"
%KEYW]		
"Per contattare l[']autore"	--	"[paragraph break]Hai bisogno di aiuto?[line break]Il mio indirizzo è [bold type]tetractys@users.sf.net[roman type]"
"Ringraziamenti"	--	"[paragraph break]Si ringraziano:[paragraph break][ringrazia-news][ringrazia-gente]"

```

[HYPER%]
To say show-HI-message:
    show HI message for hyperlink-introduction.
[%HYPER]
[KEYW%]
To say show-KI-message:
    show KI message for keyword-introduction.
%KEYW]

```

```

To say ringrazia-news:
    say "Il newsgroup italiano [bold type]it.comp.giochi.avventure.testuali[roman type]".

To say ringrazia-gente:
    say "[line break][bold type]Graham Nelson[roman type] per lo sviluppo dell[']editor
Inform7[line break][bold type]Massimo Stella[roman type] per l[']estensione 'Italian'[line break]
[bold type]Aaron Reed[roman type] per varie estensioni[line break][bold type]Emily Short[roman type]
per varie estensioni[line break][bold type]Jon Ingold[roman type] per varie estensioni[line break]
[bold type]Eric Eve[roman type] per varie estensioni"

```

```

[$%]
Use menus.

The intro menu is the Table of Basic Help Options.

To say quotation:
    say "Un cammello non prende in giro un altro cammello per le sue gobbe -- Proverbio arabo".
[%$]

```

Use menus.

The intro menu is the Table of Basic Help Options.

```

To say quotation:
    say "Un cammello non prende in giro un altro cammello per le sue gobbe -- Proverbio arabo".
[%$]

```

Chapter 5 - Status Line

Understand the command "uscite" as something new.

Include Exit Lister IT by Leonardo Boselli.

left	central	right
" [location] "	"punti: [score]/[maximum score]"	" passi: [turn count] stanze: [number of visited rooms]/[number of rooms]"

left	central	right
" [location] "	"punti: [score]/[maximum score]"	" passi: [turn count] stanze: [number of visited rooms]/[number of rooms]"

```
" [exit list]" "" ""
```

```
First when play begins:
  change standard status table to Table of No Exits Status;
  change status exit table to Table of Exits Status;
  change right alignment depth to 30.
```

```
Instead of going nowhere:
  say "Non puoi andare in quella direzione.";
  list the exits.
```

Chapter 6 - Il punteggio

The maximum score is 1000.

```
[HYPER%]
```

Chapter 7 - Azioni suggerite

Definition: a thing is takeable if it is not scenery and it is not a person and it is not part of something and it is not an enterable container and (it is a cosa-calpestable or it is not a supporter).

Definition: a thing is pocketable if it is not a cosa-ingombrante and it is not a cosa-calpestable and it is not a container.

After examining a thing (called the object):

```
  say "[fixed letter spacing]";
  if the object is worn by the player:
    change HI hyperlink text to "togli [the object]";
    print HI hyperlink;
  otherwise:
    if the object is wearable:
      change HI hyperlink text to "indossa [the object]";
      print HI hyperlink;
    if the object is carried by the player:
      change HI hyperlink text to "lascia [the object]";
      print HI hyperlink;
      if the object is edible:
        change HI hyperlink text to "annusa [the object]";
        print HI hyperlink;
        change HI hyperlink text to "assaggia [the object]";
        print HI hyperlink;
        change HI hyperlink text to "mangia [the object]";
        print HI hyperlink;
      otherwise if the object is a puzzle:
        if the object is mescolato:
          change HI hyperlink text to "risolvi [the object]";
          print HI hyperlink;
        otherwise:
          change HI hyperlink text to "mescola [the object]";
          print HI hyperlink;
      otherwise if the object is a digging-tool:
        change HI hyperlink text to "scava";
        print HI hyperlink;
      otherwise if the object is a spruzzatore:
        change HI hyperlink text to "agita [the object]";
        print HI hyperlink;
        repeat with P running through persons in the location:
          if P is not the player:
            change HI hyperlink text to "spruzza [the object] [su-
prep the P]";
            print HI hyperlink;
      otherwise if the object is the matassa di fili di lana:
        change HI hyperlink text to "intreccia la matassa";
        print HI hyperlink;
      otherwise if the object is the fune:
        if the cammello is visible and the asino is visible:
          change HI hyperlink text to "lega l'asino al cammello";
          print HI hyperlink;
      otherwise if the object is the forbici:
        if the fune is visible:
          change HI hyperlink text to "taglia la fune";
          print HI hyperlink;
        if the rosa is visible:
          change HI hyperlink text to "taglia la rosa";
          print HI hyperlink;
        if the pezzo di carne is visible:
          change HI hyperlink text to "taglia la carne";
          print HI hyperlink;
      if the player wears the grembiule and the object is pocketable:
```

```

        change HI hyperlink text to "metti [the object] in tasca";
        print HI hyperlink;
    repeat with RA running through rideable animals in the location:
        change HI hyperlink text to "metti [the object] [su-prep the RA]";
        print HI hyperlink;
    repeat with P running through persons in the location:
        if P is not the player:
            change HI hyperlink text to "dai [the object] [a-prep the P]";
            print HI hyperlink;
    repeat with S running through supporters in the location:
        if S is not the object:
            change HI hyperlink text to "metti [the object] [su-prep the
S]";

            print HI hyperlink;
            if S is a cosa-calpestable:
                change HI hyperlink text to "metti [the object] sotto
[the S]";

                print HI hyperlink;
    repeat with C running through open containers in the location:
        if C is not the object:
            change HI hyperlink text to "metti [the object] [in-prep the
C]";

            print HI hyperlink;
    otherwise if the object is takeable:
        change HI hyperlink text to "prendi [the object]";
        print HI hyperlink;
    if the object is a container:
        if the object is open:
            if the object is openable:
                change HI hyperlink text to "chiudi [the object]";
                print HI hyperlink;
            repeat with Obj running through the things carried by the player:
                change HI hyperlink text to "metti [the Obj] [in-prep the
object]";

                print HI hyperlink;
        otherwise:
            change HI hyperlink text to "apri [the object]";
            print HI hyperlink;
    if the object is enterable:
        if the object is a supporter:
            if the player is on the object:
                change HI hyperlink text to "scendi";
                print HI hyperlink;
            otherwise:
                change HI hyperlink text to "sali [su-prep the object]";
                print HI hyperlink;
            if the object is a cosa-calpestable:
                change HI hyperlink text to "tocca [the object]";
                print HI hyperlink;
                change HI hyperlink text to "guarda sotto [the object]";
                print HI hyperlink;
        otherwise if the object is a container:
            if the player is in the object:
                change HI hyperlink text to "esci";
                print HI hyperlink;
            otherwise:
                change HI hyperlink text to "entra [in-prep the object]";
                print HI hyperlink;
    otherwise if the object is a rideable animal:
        if the player is carried by the object:
            change HI hyperlink text to "scendi [da-prep the object]";
            print HI hyperlink;
        otherwise:
            change HI hyperlink text to "sali [su-prep the object]";
            print HI hyperlink;
            repeat with Obj running through the things carried by the player:
                change HI hyperlink text to "metti [the Obj] [su-prep the
object]";

                print HI hyperlink;
            change HI hyperlink text to "spingi [the object]";
            print HI hyperlink;
            repeat with S running through the things carried by the player:
                change HI hyperlink text to "dai [the S] [a-prep the object]";
                print HI hyperlink;
    otherwise if the object is the vecchio saggio:
        change HI hyperlink text to "svegliata [the object]";
        print HI hyperlink;
        change HI hyperlink text to "parla [a-prep the object]";
        print HI hyperlink;

```

```

otherwise if the object is a person:
    repeat with S running through the things carried by the player:
        change HI hyperlink text to "dai [the S] [a-prep the object]";
        print HI hyperlink;
otherwise if the object is the rosa:
    change HI hyperlink text to "annusa [the object]";
    print HI hyperlink;
    if the player carries the forbici:
        change HI hyperlink text to "taglia [the object]";
        print HI hyperlink;
if the object is the porta di pietra and the porta di pietra is stregata:
    change HI hyperlink text to "apriti sesamo";
    print HI hyperlink;
otherwise if the object is the fune:
    if the player carries the forbici:
        change HI hyperlink text to "taglia [the object]";
        print HI hyperlink;
otherwise if the object is the lungo remo:
    change HI hyperlink text to "spingi remo";
    print HI hyperlink;
    change HI hyperlink text to "tira remo";
    print HI hyperlink;
otherwise if the object is the striscioline di carne:
    repeat with Obj running through persons in the location:
        if the Obj is not the player:
            change HI hyperlink text to "lancia [the object] [a-prep the
Obj]";
            print HI hyperlink;
say "[roman type][no line break]";
continue the action.
[%HYPER]

```

Part II - Multimedia

```

[$%]
Include Flexible Windows by Jon Ingold.
The main-window has back-colour g-light-grey;

```

Section 1 - Files

```

Figure of titolo is the file "titolo.jpg".
Figure of deserto-di-sabbia is the file "deserto-di-sabbia.jpg".
Figure of margini-dell-oasi is the file "margini-dell-oasi.jpg".
Figure of oasi is the file "oasi.jpg".
Figure of serraglio is the file "serraglio.jpg".
Figure of palmeto-dell-oasi is the file "palmeto-dell-oasi.jpg".
Figure of filatoio-dell-oasi is the file "filatoio-dell-oasi.jpg".
Figure of magazzino is the file "magazzino.jpg".
Figure of giardino-incantato is the file "giardino-incantato.jpg".
Figure of stalla is the file "stalla.jpg".
Figure of sorgente-dell-oasi is the file "sorgente-dell-oasi.jpg".
Figure of grotta-dei-quaranta-ladroni is the file "grotta-dei-quaranta-ladroni.jpg".
Figure of accesso-alla-grotta is the file "accesso-alla-grotta.jpg".
Figure of gola-tra-le-montagne is the file "gola-tra-le-montagne.jpg".
Figure of cima-della-montagna is the file "cima-della-montagna.jpg".
Figure of valle-delle-tombe is the file "valle-delle-tombe.jpg".
Figure of accampamento-dei-predoni is the file "accampamento-dei-predoni.jpg".
Figure of riva-sinistra-del-fiume is the file "riva-sinistra-del-fiume.jpg".
Figure of grande-fiume is the file "grande-fiume.jpg".
Figure of via-delle-carovane is the file "via-delle-carovane.jpg".
Figure of aperta-savana is the file "aperta-savana.jpg".
Figure of villaggio-degli-agricoltori is the file "villaggio-degli-agricoltori.jpg".
Figure of vigneti-delle-colline is the file "vigneti-delle-colline.jpg".
Figure of campi-di-grano is the file "campi-di-grano.jpg".
Figure of bosco-di-acacie is the file "bosco-di-acacie.jpg".
Figure of deserto-di-sassi is the file "deserto-di-sassi.jpg".
Figure of inferno-di-sassi is the file "inferno-di-sassi.jpg".
Figure of deserto-di-pietre is the file "deserto-di-pietre.jpg".
Figure of deserto-di-rocce is the file "deserto-di-rocce.jpg".

Figure of uomo-e-orme is the file "uomo-e-orme-V.jpg".
Figure of uomo-e-dario-1 is the file "uomo-e-dario-1-V.jpg".
Figure of uomo-e-dario-2 is the file "uomo-e-dario-2-V.jpg".
Figure of asino is the file "asino-V.jpg".
Figure of avvoltoio is the file "avvoltoio-V.jpg".
Figure of cascata is the file "cascata-V.jpg".
Figure of fune is the file "fune-V.jpg".
Figure of dario-in-piedi is the file "dario-1-V.jpg".
Figure of dario-seduto is the file "dario-2-V.jpg".

```

Figure of entrata-grotta is the file "entrata-grotta-V.jpg".
 Figure of formicaio is the file "formicaio-V.jpg".
 Figure of filati is the file "filati-V.jpg".
 Figure of forziere is the file "forziere-V.jpg".
 Figure of otri is the file "otri-V.jpg".
 Figure of grano is the file "grano-V.jpg".
 Figure of leone-in-piedi is the file "leone-in-piedi-V.jpg".
 Figure of leone-sdraiato is the file "leone-sdraiato-V.jpg".
 Figure of matasse is the file "matasse-V.jpg".
 Figure of orda-di-predoni is the file "orda-di-predoni-V.jpg".
 Figure of pappagallo is the file "pappagallo-V.jpg".
 Figure of pesce is the file "pesce-V.jpg".
 Figure of rosa is the file "rosa-V.jpg".
 Figure of tappeti is the file "tappeti-V.jpg".
 Figure of uva is the file "uva-V.jpg".
 Figure of zattera is the file "zattera-V.jpg".
 Figure of palme-1 is the file "palme-1-V.jpg".
 Figure of palme-2 is the file "palme-2-V.jpg".
 Figure of palme-3 is the file "palme-3-V.jpg".
 Figure of forbici is the file "cesoie-V.jpg".
 Figure of spray is the file "spray-V.jpg".
 Figure of arnia is the file "arnia-V.jpg".
 Figure of pala is the file "pala-V.jpg".
 Figure of favo is the file "favo-V.jpg".
 Figure of tino is the file "tino-V.jpg".

Sound of colonna sonora is the file "daydreaming.ogg".
 Sound of applauso is the file "applauso.ogg".
 Sound of avvoltoio-attacco is the file "aquila-attacco.ogg".
 Sound of avvoltoio-in-volo is the file "aquila-in-volo.ogg".
 Sound of asino is the file "asino.ogg".
 Sound of caduta is the file "caduta.ogg".
 Sound of cammello is the file "cammello.ogg".
 Sound of cicale is the file "cicale.ogg".
 Sound of leone is the file "leone.ogg".
 Sound of onde is the file "onde.ogg".
 Sound of tuffo is the file "tuffo.ogg".

Section 2 - Illustrazioni delle stanze

A room has a figure-name called top-illustration.

The image-setting rule is listed in the carry out looking rules.

This is the image-setting rule:

```
show-top the top-illustration of the location;
```

Section 3 - Immagini e suoni (for Glulx only)

First when play begins:

```
play sound of colonna sonora.
```

The top-window is a g-window spawned by the main-window.

The type is g-graphics. The position is g-placeabove. The scale method is g-fixed-size. The measurement is 150. The back-colour is g-black. The drawing rule is the draw top-window rule. The current top-window figure is a figure-name that varies.

This is the draw top-window rule:

```
if top-window is g-unpresent, rule fails;
move focus to top-window, clearing the window;
draw the current top-window figure in top-window with width 600 and height 150;
return to main screen.
```

The corner-window is a g-window spawned by the top-window.

The type is g-graphics. The position is g-placeleft. The scale method is g-fixed-size. The measurement is 150. The back-colour is g-black.

The drawing rule is the draw corner-window rule.

This is the draw corner-window rule:

```
if corner-window is g-unpresent, rule fails;
move focus to corner-window, clearing the window;
draw the figure of titolo in corner-window with width 150 and height 150;
return to main screen.
```

The side-window is a g-window spawned by the main-window.

The type is g-graphics. The position is g-placeleft. The scale method is g-fixed-size. The measurement is 150. The back-colour is g-black. The drawing rule is the draw side-window rule. The current side-window figure is a figure-name that varies.

This is the draw side-window rule:

```
if side-window is g-unpresent, rule fails;
move focus to side-window, clearing the window;
```

```
draw the current side-window figure in side-window with width 150 and height 300;
return to main screen.
```

```
To show-side (figure - a figure-name):
change the current side-window figure to figure;
follow the draw side-window rule.
```

```
To show-top (figure - a figure-name):
change the current top-window figure to figure;
follow the draw top-window rule.
```

```
To draw the (f - a figure-name) in (g - a g-window) with width (w - a number) and height (h - a
number):
(- glk_image_draw_scaled({g}.ref_number, {f}, 0, 0, {w}, {h}); -) .
```

```
When play begins:
open up the top-window;
open up the corner-window;
open up the side-window.
```

Section 4 - nessuna immagine o suono (for Z-machine only)

```
To show-side (figure - a figure-name):
do nothing.
```

```
To show-top (figure - a figure-name):
do nothing.
```

```
[%$]
```

Part III - I proverbi del vecchio saggio

Chapter 1 - Risposte

```
Before saying hello to the vecchio saggio:
change the current interlocutor to the noun;
try saggio-talking to instead.
```

Saggio-telling about is an action applying to a topic.
 Saggio-asking about is an action applying to a topic.
 Saggio-asking for is an action applying to one thing.
 Saggio-talking to is an action applying to nothing.

```
Understand "parla vecchio/saggio" or "parla [a-art] vecchio/saggio" or "parla [con-art]
vecchio/saggio" as saggio-talking to.
Understand "parla vecchio/saggio [di-art] [text]" or "parla [a-art] vecchio/saggio [di-art] [text]"
or "parla [con-art] vecchio/saggio [di-art] [text]" as saggio-telling about.
Understand "parla vecchio saggio [di-art] [text]" or "parla [a-art] vecchio saggio [di-art] [text]"
or "parla [con-art] vecchio saggio [di-art] [text]" as saggio-telling about.
Understand "chiedi vecchio/saggio [di-art] [text]" as saggio-asking about.
Understand "chiedi vecchio saggio [di-art] [text]" as saggio-asking about.
Understand "chiedi [a-art] vecchio/saggio [di-art] [text]" as saggio-asking about.
Understand "chiedi [a-art] vecchio saggio [di-art] [text]" as saggio-asking about.
Understand "chiedi [a-art] vecchio/saggio [something]" or "chiedi [a-art] vecchio/saggio [art-det]
[something]" as saggio-asking for.
Understand "chiedi [a-art] vecchio saggio [something]" or "chiedi [a-art] vecchio saggio [art-det]
[something]" as saggio-asking for.
```

```
Check saggio-asking for:
if the saggio is not visible, say "[no-saggio]" instead.
```

```
Carry out saggio-asking for:
say "Il saggio si sveglia di soprassalto e dice: 'Mi chiedi [the noun]... [the noun]! [The
description of the noun] Ma sono troppo vecchio per muovermi.' [line break]";
choose a random row in the Table of Proverbs;
say "Inoltre aggiunge: '[italic type][proverb entry][roman type]'. [run paragraph on]";
conclude proverb.
```

```
Check saggio-asking about:
if the saggio is not visible, say "[no-saggio]" instead.
```

```
Carry out saggio-asking about:
say "Il saggio si scuote dal suo torpore e ripete tra sé: '[italic type]Mi chiedi di [the
topic understood]... [the topic understood]?[roman type]'";
declare proverb on "[the the topic understood]".
```

```
Check saggio-telling about:
if the saggio is not visible, say "[no-saggio]" instead.
```

```
Carry out saggio-telling about:
say "Il saggio si sveglia e rimugina: '[italic type]Mi parli di [the topic understood]...
[the topic understood]?[roman type]'";
declare proverb on "[the topic understood]".
```

```
Check saggio-talking to:
if the saggio is not visible, say "[no-saggio]" instead.
```

```

Carry out saggio-talking to:
    say "Il saggio si sveglia e ti guarda meravigliato. ";
declare proverb.
Instead of waking the vecchio saggio:
    say "Il saggio si sveglia e ti guarda stupito. ";
declare proverb.

To say no-saggio:
    say "Non c[']è nessun vecchio saggio in vista."

```

Chapter 2 - Il proverbio

```

To introduce proverb:
    say "[line break][one of]Dopo aver riflettuto per lunghi istanti,[or]Si stropiccia gli
occhi, ti guarda con attenzione poi[or]Per qualche attimo si tira la barba, poi[or]Riflette a lungo,
sospira e, infine,[or]Dopo aver attentamente meditato,[at random]".

```

```

To conclude proverb:
    say commento-su-vestiario;
    say "[line break][one of]Un sorriso compiaciuto si disegna sul suo volto[or]Senza aspettare
una tua risposta reclina il capo[or]Con la stessa rapidità con cui si era svegliato, chiude gli
occhi[or]Si copre gli occhi con la mano[or]Dopo aver manifestato la sua immensa saggezza,
sospira[or]Dopo aver sciorinato le sue perle di saggezza, tace[at random] e ";
    say "[one of]subito si assopisce di nuovo[or]si addormenta[or]torna nel mondo dei
sogni[or]ricomincia a russare[or]riprende il sonno del giusto[at random].".

```

```

To say commento-su-vestiario:
    if the player is nudo, say "[line break]Dopo aver riflettuto ancora aggiunge: 'Ti suggerisco
anche di trovare dei vestiti, perché [one of]forse tu non l[']hai notato, ma sei nudo come un
verme... e non è un bel vedere[or]fa sicuramente caldo, ma tu stai esagerando nel prendere
aria[or]ci sono mariti gelosi qui in giro che potrebbero temere il confronto[or]non tutti apprezzano
l[']esibizione disinibita del proprio corpo[or]la vista delle tue nudità potrebbe turbare i
cammelli[or]gli avvoltoi sono sempre in cerca di carne nuda[or]la vista del tuo corpo nudo può far
appassire i fiori[or]potresti lanciare una nuova moda e sono troppo vecchio per combinarmi
così[or]gli uomini invidiosi possono diventare violenti e passare alle vie di fatto[or]potresti
prendere un raffreddore[or]il sole brucia ed è meglio coprire le parti delicate[at random][commento-
su-sandali].'[run paragraph on]"

```

```

To say commento-su-sandali:
    if the player wears a paio di sandali, say ". [one of]Un paio di sandali, anche se bello
come il tuo, non basta[or]Indossando solo un paio di sandali, fai poca strada[or]Indossando solo un
paio... Ehi! Vedo che abbiamo gli stessi gusti in fatto di sandali[or]Un paio di sandali non... Ehi!
Hai davvero un bel paio di sandali, sembrano i miei[at random]".

```

```

To declare proverb on (object - a text):
    choose a random row in the Table of Proverbs;
    repeat with N running from 1 to the number of rows in the Table of Proverbs
    begin;
        if the proverb in row N of the Table of Proverbs matches the text "[object]"
        begin;
            choose row N in the Table of Proverbs;
            break;
        end if;
    end repeat;
    introduce proverb;
    say " proclama: '[italic type][proverb entry][roman type].'[run paragraph on]";
    conclude proverb.

```

```

To declare proverb:
    choose a random row in the Table of Proverbs;
    introduce proverb;
    say " proclama: '[italic type][proverb entry][roman type],"';
    choose a random row in the Table of Proverbs;
    say " poi continua dicendo: '[italic type][proverb entry][roman type]';";
    choose a random row in the Table of Proverbs;
    say " ed infine conclude con: '[italic type][proverb entry][roman type].'[no line break]";
    conclude proverb.

```

Chapter 3 - La lista di proverbi

```

Table of Proverbs
proverb
"Ciò che è passato è andato"
"L[']ignorante è nemico di se stesso"
"Una sola mano non può applaudire"
"Non spingere l[']ubriaco, casca da solo"
"Scegli prima il tuo vicino e poi la casa"
"Non dire due parole se una è sufficiente"
"Colui che veste gli abiti degli altri è nudo"

```

"Meno il cuore è nobile, più la testa è eretta"
 "Chi vive vede molto, chi viaggia vede di più"
 "La vita è un ramo di palma piegato dai venti"
 "Un beneficio rinfacciato vale quanto un'offesa"
 "Non puoi cavare farina da un sacco di carbone"
 "Si può vivere senza fratelli, ma non senza amici"
 "Esamina ciò che viene detto, non colui che parla"
 "Il frutto della pace è appeso all[']albero del silenzio"
 "Chi scava un fosso con cattiveria, ci finisce dentro"
 "Un uomo può valerne cento e cento non valerne uno"
 "L[']avido è colui che, se ti cade un occhio, dice che è suo"
 "Chi vuole il miele deve sopportare le punture delle api"
 "Quando c[']è una meta, anche il deserto diventa strada"
 "Allunga il passo secondo la grandezza del tuo tappeto"
 "La stessa acqua non scorre due volte nello stesso fiume"
 "Un dattero non cade mai troppo lontano dal suo albero"
 "Prima di tagliare con la forbice devi pensare cento volte"
 "L[']avido è colui che pretende un pettine, anche se è calvo"
 "Meglio passare una notte con la collera che con il rimorso"
 "Un nobile cuore non può godere d[']una fortuna tolta ad altri"
 "Le tempeste dell[']anima sono peggiori delle tempeste di sabbia"
 "Han dato una rosa a un asino, invece di annusarla l[']ha mangiata"
 "La menzogna fa bere una prima volta. Non lascia bere una seconda"
 "Un cammello non prende in giro un altro cammello per le sue gobbe"
 "Se ti fermi ogni volta che un cane abbaia, non finirai mai la tua strada"
 "Delle tribù gli uomini sono la lana, ma sono le donne a tesserne la trama"
 "Chi corre sempre, saprà sempre meno cose di colui che sta calmo e riflette"
 "Quando si trova un amico nuovo, non bisogna dimenticare quello vecchio"
 "Il miglior discorso è quello che con poche parole riesce ad essere significativo"
 "Affida il tuo cammello alla provvidenza di Dio, ma legalo prima ad un albero"
 "E['] più facile proteggersi i piedi con i sandali che ricoprire di tappeti tutta la terra"
 "Sotto il sole del deserto il cammelliere fa i suoi progetti, ma li fa anche il cammello"
 "Se tutte le formiche si riunissero, finirebbero con il sopraffare i più formidabili leoni"
 "Una volta intrecciati, anche i capelli più sottili divengono più forti di una catena di ferro"
 "Mettiti in cammino anche se l[']ora non ti piace. Quando arriverai, l[']ora ti sarà comunque gradita"
 "Il genere umano si divide in tre classi: gli inamovibili, quelli che sono mossi, e quelli che muovono"
 "Non essere troppo molle perchè ti schiacceranno, non essere troppo duro perchè ti faranno a pezzi"
 "Incarica un guardiano di sorvegliare i tuoi gioielli, ma non incaricare mai nessuno di sorvegliare i tuoi segreti"
 "Chi ti ama amalo e fai che il tuo amore per lui sia sincero, mentre chi ti odia non odiarlo, lascialo vivere nell[']inferno e nel fuoco"
 "Non biasimate un re che si umili per amore, poichè abbassarsi per amore è segno di potenza ed equivale ad essere due volte re"
 "L[']occhio che non ti ha visto, è come se non avesse mai visto niente. Il cuore che non ti ha amato è come se non avesse mai vissuto"
 "Un bicchiere di tè è niente, due bicchieri è da poveri, tre vanno bene, quattro fanno piacere, cinque sono proibiti, sei sono meglio di tre"
 "La piccola formica non teme mai la carestia; il leone, nonostante i suoi denti aguzzi e gli artigli acuminati, non sempre trova da mangiare"
 "Ai cattivi come ai buoni, fai sempre la carità. Facendola ai buoni, aggiungi merito alle tue virtù; facendola ai cattivi, eviti il pericolo d[']ingannarti"
 "Onesto è colui che cambia il proprio pensiero per accordarlo alla verità. Disonesto è colui che cambia la verità per accordarla al proprio pensiero"
 "Colui che non sa, e non sa di non sapere, è uno sciocco: evitalo. Colui che non sa, e sa di non sapere, è un ignorante: istruiscilo. Colui che sa, e non sa di sapere, è addormentato: sveglialo. Colui che sa, e sa di sapere, è un saggio: seguilo"

To say elenca-proverbi:

```
say "[paragraph break][bold type]I detti del vecchio saggio[roman type][paragraph break]";
repeat with P running from 1 to the number of rows in the Table of Proverbs:
  say "[P]) [italic type][proverb in row P of the Table of Proverbs].[roman type][line
break]";
  if the remainder after dividing P by 10 is 0:
    say "[line break](premi un tasto per continuare)[paragraph break]";
    wait for any key.
```

Part IV - L'avventura

Use full-length room descriptions.

Section 1 - I tipi

The luogo-lontano is a kind of room.

The luogo-molto-lontano is a kind of luogo-lontano.

The cosa-ingombrante is a kind of thing. The bulk of a cosa-ingombrante is 50.

The cosa-calpestable is a kind of supporter. A cosa-calpestable is usually enterable. A cosa-calpestable is usually portable. The bulk of a cosa-calpestable is 25.

A thing can be examined or unexamined. A thing is usually unexamined.

After examining a thing (called the object):
 now the object is examined;
 continue the action.

Section 2 - Il giocatore

To decide whether the player is nudo:
 if the player is wearing the grembiule, decide no;
 otherwise decide yes.

To decide whether the player is not nudo:
 if the player is wearing the grembiule, decide yes;
 otherwise decide no.

Before taking something (called the preso):
 if the number of things carried by the player is greater than 1, say "Non puoi prendere [the preso], perché hai già [list of things carried by the player with indefinite articles] in mano."
 instead.

The description of the player is "Nonostante la brutta avventura, sei sempre lo stesso[if player is nudo]: solo un po['] troppo nudo per andartene in giro impunemente[end if]."

Section 3 - Discussioni

Check hailing (this is the hail-not-animals rule):
 if the current interlocutor is a visible person then
 say "Stai già parlando [con-prep the current interlocutor]." instead;
 change the noun to a random visible men who is not the player;
 if the noun is not a person then
 change the noun to a random visible women who is not the player;
 if the noun is a person then
 say "(rivolgendoti [a-prep the noun])";
 otherwise
 say "Qui ci sei solo tu." instead.

The hail-not-animals rule is listed instead of the check what's being hailed rule in the check hailing rules.

Understand "argo" as listing suggested topics.

A thing can be discussable or speakable or unspeakable. A thing is usually unspeakable.

Instead of examining a speakable thing (called the argument) when the current interlocutor is a person:
 if the argument is listed in the ask-suggestions of the node of the current interlocutor:
 try quizzing the current interlocutor about the argument;
 otherwise if the argument is listed in the tell-suggestions of the node of the current interlocutor:
 try informing the current interlocutor about the argument;
 otherwise:
 continue the action.

Instead of examining a discussable thing (called the argument):
 if the current interlocutor is a person:
 if the argument is listed in the ask-suggestions of the node of the current interlocutor:
 try quizzing the current interlocutor about the argument;
 otherwise if the argument is listed in the tell-suggestions of the node of the current interlocutor:
 try informing the current interlocutor about the argument;
 otherwise:
 say "[arg-discusso argument]";
 otherwise:
 say "[arg-discusso argument]."

To say arg-discusso (argument - a discussable thing):
 say "[The argument] [if the argument is plural-named]sono solo[otherwise]è solo un[end if] argomento di discussione."

Chapter 1 - Il deserto di sabbia

When play begins:
 [%] show-top figure of deserto-di-sabbia; [%]
 say mostra-uomo-e-orme;
 say "Lo avevi sentito raccontare tante volte, ma ora è capitato a te.[paragraph break]Tutti

i mercanti più anziani, alla luce incerta dei fuochi dei serragli, narravano storie simili e, con l'aiuto di abbondanti bevute a base di latte di cammella fermentato, non lesinavano sui particolari raccapriccianti... ma questa volta è capitato a te e fai ancora fatica a rendertene conto. Chissà quanto sei rimasto privo di conoscenza. La caduta dal cammello ti ha lasciato un brutto ricordo sulla nuca. Lo senti pulsare, lo cerchi con la mano e il dolore si fa più intenso. Ti tornano in mente la fuga attraverso le dune, le urla dei predoni, gli spari dei fucili ed i grugniti dei cammelli, poi l'improvvisa caduta ed infine... più nulla. Ti guardi intorno e non vedi che sabbia, dune di sabbia a perdita d'occhio. I predoni ti hanno portato via tutto quello che possedevi, ma ti hanno lasciato la vita o, forse, non hanno perso tempo a togliertela: ci penserà il deserto. Ti schiaffeggia un vento pesante, caldo, carico di polvere. Il sole già alto domina il cielo e ha scacciato il gelo della notte. Presto diventerà ancora più soffocante. Hai bisogno di trovare un riparo e acqua... tanta acqua! Scrivi *Scrivi 'info'* per le istruzioni o *'argo'* quando parli con qualcuno.

```
To say mostra-uomo-e-orme:
    [%] show-side figure of uomo-e-orme; [%]
    do nothing.
```

The deserto di sabbia is a room. "Intorno a te riesci a vedere solo sabbia, dune di sabbia in tutte le direzioni. La via di casa è sicuramente a [est] se non sei trasportato, ma è troppo lontana per arrivarci a piedi. A [nord] puoi vedere una distesa di sassi, mentre a [sud] una sterminata pietraia. Invece a [ovest] scorgi in lontananza le palme di un'oasi." The top-illustration is figure of deserto-di-sabbia.

The sole is a backdrop in the deserto di sabbia. The description is "[sole-a-picco]".
The cielo is a backdrop in the deserto di sabbia. The description is "[sole-a-picco]".

```
To say sole-a-picco:
    say "Il cielo è sereno, di un azzurro intenso. Completamente sereno. Non si vede nessuna
    nuvola all'orizzonte che possa minacciare il sole a picco, un sole che picchia violente martellate
    di luce e calore."
```

The dune di sabbia is a female, plural-named backdrop in the deserto di sabbia. The description is "Dune... dune di sabbia ovunque, a perdita d'occhio."
The via di casa is a female backdrop in the deserto di sabbia. The description is "La via di casa: così inaccessibile, ma allo stesso tempo così vicina. La via delle carovane inizia ad est, oltre il grande fiume."
The distesa di sassi is a female, plural-named backdrop in the deserto di sabbia. The description is "Una distesa di sassi, che si estende immensa fino all'orizzonte, ricolma di ciottoli tondi e ben levigati. Un tempo doveva essere il fondo di un lago o un mare interno, mentre ora è ridotta ad un catino infuocato."
The sterminata pietraia is a female backdrop in the deserto di sabbia. The description is "Una sterminata pietraia, ricolma di pietre aguzze: l'ideale per tormentare il lento cammino dei cammelli."
The palme is a female, plural-named backdrop in the deserto di sabbia. The description is "[mostra-palme]Palme rigogliose che traggono la loro forza dal sole. La vita trionfa ovunque."

```
To say mostra-palme:
    say "[one of][mostra-palme-1][or][mostra-palme-2][or][mostra-palme-3][at random]".
```

```
To say mostra-palme-1:
    [%] show-side figure of palme-1; [%]
    do nothing.
```

```
To say mostra-palme-2:
    [%] show-side figure of palme-2; [%]
    do nothing.
```

```
To say mostra-palme-3:
    [%] show-side figure of palme-3; [%]
    do nothing.
```

```
Before going to the deserto di sabbia:
    move the avvoltoio to the deserto di sabbia;
    continue the action.
```

```
After going from the deserto di sabbia when the avvoltoio is in the deserto di sabbia:
    if the avvoltoio is assetato:
        now the avvoltoio is not assetato;
        move the avvoltoio to the sorgente;
    otherwise:
        move the avvoltoio to the cima della montagna;
    continue the action.
```

Chapter 2 - L'oasi

Section 1 - Ai margini dell'oasi

The margini dell'oasi is a plural-named room. It is west of the deserto di sabbia. "Sei ai margini

dell[']oasi. Intorno a te vedi numerose [s]palme[x]. La fresca brezza che proviene dal centro dell'oasi profuma di dolci datteri e di acqua dissetante.[line break]Andando verso [d]ovest[x] entri all[']interno dell[']oasi, mentre ad [d]est[x] torni al deserto di [s]sabbia[x]." [%\$] The top-illustration is figure of margini-dell'oasi. [%\$]

The cielo is a backdrop in the margini dell'oasi.
The palme is a female, plural-named backdrop in the margini dell'oasi.
The dune di sabbia is a female, plural-named backdrop in the margini dell'oasi.

Section 2 - Il cammello

The gobba is a female supporter. It is part of the cammello. The description is "[mostra-dario-seduto]E['] proprio vero: i dromedari hanno una sola gobba!"
Understand "gobbe" as gobba.

To say mostra-dario-seduto:
[%\$] show-side figure of dario-seduto; [%\$]
[%\$] play sound of cammello; [%\$]
do nothing.

To say mostra-dario-in-piedi:
[%\$] show-side figure of dario-in-piedi; [%\$]
[%\$] play sound of cammello; [%\$]
do nothing.

To say Nome-cammello:
say "[if targhetta is unexamined]Il [o]cammello[x][otherwise]Dario il [o]dromedario[x][end if]".

To say nome-cammello:
say "[if targhetta is unexamined]il [o]cammello[x][otherwise]Dario il [o]dromedario[x][end if]".

The cammello is a rideable animal. It is in the serraglio. "[if the targhetta is unexamined]C[']è un [o]cammello[x] molto vecchio, probabilmente abbandonato[otherwise]Vedi [nome-cammello][end if]. Sembra molto stanco, ma potrebbe avere la forza di fare ancora qualche passo.[if the targhetta is unexamined] Che strano! Porta una [o]targhetta[x].[end if]". The description of the cammello is "[mostra-dario-in-piedi][Nome-cammello] non è per niente in forma[if the cammello is affamato] ed è affamato[end if][if the number of things on the gobba is not zero]. Porta [list of things on the gobba with indefinite articles] sulla gobba[end if].[if the targhetta is unexamined] Ha una [o]targhetta[x].[end if]". The printed name is "[if targhetta is unexamined]cammello[otherwise]Dario[end if]".

The targhetta is a female thing part of the cammello. The description is "La targhetta dice:[line break]'[italic type]Questo non è un cammello, è un dromedario![line break]I cammelli hanno due gobbe! Il suo nome è Dario.[line break]Abbine cura e ti porterà molto lontano.[roman type]'[paragraph break]'Che pignoleria!', pensi scocciato, 'Queste stupide bestie con una gobba le ho sempre chiamate cammelli e non comincerò proprio adesso a chiamarle dromedari!'".

After examining the targhetta for the first time:
award 5 points;
continue the action.

Understand "dario" or "dromedario" as cammello.

A rideable animal can be affamato or sazio. A rideable animal is usually affamato.

Instead of mounting a cammello when the cammello is affamato:
say "[mostra-dario-seduto][Nome-cammello] non ci pensa minimamente a farti salire e ti allontana scalciano e grugnendo. Sembra affamato e dovresti provare a fartelo amico in qualche modo."

Instead of putting something (called the carico) on the cammello:
try putting the carico on the gobba.

Instead of mounting a cammello when the cammello is sazio and the sella per dromedari is not on the gobba:
say "[mostra-dario-seduto][Nome-cammello] non ci pensa proprio a farti cavalcare a pelo, vista la tua tenuta [if the player is not nudo]semi-[end if]adamitica. Dovresti trovare una sella."; now the sella per cammelli is in the serraglio.

Instead of mounting a cammello when the cammello is sazio and the sella per dromedari is on the gobba and the player carries something:
say "[mostra-dario-in-piedi]Non riesci a salire perché hai le mani impegnate da [the list of things carried by the player with indefinite articles]. Per reggere le redini [if targhetta is unexamined]del [o]cammello[x][otherwise]di Dario il [o]dromedario[x][end if] devi avere le mani libere."

The sella per cammelli is a female cosa-ingombrante. The keyword is "sella". "Vedi una bellissima

[o]sella per cammelli[x], tutta istoriata, appoggiata sul suo supporto." The description is "La sella è molto bella e sembra anche comoda."

Instead of putting the sella per cammelli on the cammello:

```
now the sella per dromedari is in the serraglio;
say mostra-dario-in-piedi;
if the targhetta is unexamined:
    say "Il cammello rifiuta la sella e scarta furiosamente. Pensi tra te e te: 'Perché
deve essere così difficile? Se solo ci fossero le istruzioni!';"
otherwise:
    say "Dario il dromedario si impenna e rifiuta la sella. Ti chiedi nel profondo: 'Per
quale ragione esistenziale questo maledetto dromedario non accetta una [o]sella per cammelli[x]?'"
```

The sella per dromedari is a female cosa-ingombrante. The keyword is "sella". "C[']è anche una [o]sella per dromedari[x] usata in un angolo, così vecchia e consumata che ad una prima occhiata ti era sfuggita." The description is "La sella è vecchia, ma soprattutto scomoda."

After putting the sella per dromedari on the gobba for the first time:

```
award 50 points;
continue the action.
```

Instead of kissing a rideable animal:

```
say "Ora le ho proprio viste tutte!".
```

Instead of kissing a rideable animal for the second time:

```
say "Ci hai proprio preso gusto!".
```

Instead of kissing a rideable animal for the third time:

```
say "Non posso più guardare!".
```

Instead of kissing a rideable animal for more than the third time:

```
say "Non sto guardando, mi sto coprendo gli occhi!".
```

Instead of giving the dattero to the cammello:

```
now the cammello is not affamato;
award 10 points;
remove the dattero from play;
say "[mostra-dario-in-piedi]Il cammello afferra avidamente il dattero con la lunga lingua e
grugnisce riconoscente."
```

The check-not-animal rule is listed instead of the can't take people's possessions rule in the check taking rules.

Check an actor taking (this is the check-not-animal rule):

```
let the local ceiling be the common ancestor of the actor with the noun;
let H be the not-counting-parts holder of the noun;
while H is not nothing and H is not the local ceiling:
    if H is a person and H is not an animal, stop the action with library message taking
action number 6 for H;
let H be the not-counting-parts holder of H;
```

After going somewhere by a rideable animal (called the mounted):

```
say mostra-uomo-e-dario-1;
if the asino is tied:
    if the forziere is on the dorso:
        say mostra-uomo-e-dario-2;
        move the asino to room gone to;
    otherwise:
        now the asino is untied;
        now the fune is in the room gone from;
        say "[mostra-asino]L[']asino con uno strattone si libera dalla fune e non ti
segue. Probabilmente la soma che ha sul dorso non è abbastanza pesante per i suoi gusti.";
        say "[The mounted] trotta nella direzione [di-prep the room gone to][if the asino is tied] e
l[']asino lo segue diligentemente[end if].";
continue the action.
```

After going somewhere when the player is not carried by a rideable animal:

```
say mostra-uomo-e-orme;
continue the action.
```

Instead of going to a luogo-lontano when the player is not carried by a rideable animal:

```
[$%] show-top the top-illustration of the room gone to; [%$]
say mostra-uomo-e-orme;
say "[The room gone to] è troppo distante. Moriresti di sete. Devi usare una cavalcatura."
```

Instead of going from a luogo-lontano when the player is not carried by a rideable animal:

```
if (a rideable animal is in the location) or (the room gone to is the deserto di sabbia):
    [$%]
    show-top the top-illustration of the room gone to;
    [%$]
    say mostra-uomo-e-orme;
    say "[The room gone to] è troppo distante. Moriresti di sete. Non devi abbandonare la
```

```
cavalcatura.";
    otherwise:
        continue the action.
```

Instead of going to a luogo-molto-lontano when the player is not carried by the cammello or the otri pieni are not on the gobba:

```
[$%] show-top the top-illustration of the room gone to; [%$]
say mostra-uomo-e-orme;
say "[The room gone to] è troppo caldo. Devi portare molta acqua con te [if the player is
carried by a rideable animal (called the mounted)]altrimenti tu e [the mounted] morirete[otherwise]e
trovare una cavalcatura altrimenti morirai[end if] di sete."
```

Instead of going from a luogo-molto-lontano when the player is not carried by the cammello or the otri pieni are not on the gobba:

```
[$%] show-top the top-illustration of the room gone to; [%$]
say mostra-uomo-e-dario-1;
say "[The room gone to] è veramente troppo lontano. Devi portare molta acqua con te [if the
player is carried by a rideable animal (called the mounted)]altrimenti tu e [the mounted]
morirete[otherwise]e non devi abbandonare la cavalcatura altrimenti morirai[end if] di sete."
```

Section 3 - L'oasi

The oasi is a female room. It is west of the margini dell'oasi. "Sei al centro dell[']oasi. Puoi vedere molte [s]palme[x] tutt[']intorno e la loro ombra ti ristora.[line break]A [d]nord[x] c[']è un [s]filatoio[x] dove le donne del villaggio si raccolgono per filare e tessere; a [d]sud[x] c[']è il [s]serraglio[x], dove invece si radunano i mercanti di passaggio; ad [d]ovest[x] vedi un fitto [s]palmeto[x]: sono tutte palme da dattero; infine, ad [d]est[x] torni ai margini dell[']oasi." [%\$]
The top-illustration is figure of oasi. [%\$]

The edificio del filatoio is a backdrop in the oasi. The description is "Una bassa costruzione in legno, circondata dalle palme dell[']oasi."

The recinto del serraglio is a backdrop in the oasi. The description is "Un recinto che separa un[']area in terra battuta dal resto dell[']oasi."

The palmeto di datteri is a backdrop in the oasi. The description is "[mostra-palme]Un fitto palmeto. La produzione di datteri deve essere ingente, ma forse non in questa stagione."

The palme is a female, plural-named backdrop in the oasi.

Section 4 - Il dattero

The dattero is an edible thing. The dattero is in the palmeto dell'oasi. "C[']è un [o]dattero[x] per terra. Solo uno! Fanne buon uso." The description is "[mostra-palme]Sembra veramente gustoso."

After eating the dattero for the first time:

```
say "Molto buono! Ma non era questo che intendevo suggerendo: 'Fanne buon uso!'[paragraph
break]";
if the cammello is affamato, say "Ti conviene cercarne altri perché ti serviranno.";
continue the action.
```

After eating the dattero for more than one time:

```
say "Ma allora sei recidivo! Il dattero aveva un altro scopo.[paragraph break]";
if the cammello is affamato, say "Cercane un altro. Ti servirà!";
continue the action.
```

After going to the palmeto when the dattero is off-stage and the cammello is affamato:

```
say "[mostra-palme]Una raffica di vento fa cadere un dattero da una palma. A quanto pare,
per fortuna, ce n[']era ancora uno.";
move the dattero to the palmeto;
continue the action.
```

Instead of smelling the dattero:

```
say "Un profumo che ricorda il miele!"
```

Instead of tasting the dattero:

```
say "Ha lo stesso gusto del miele!"
```

Section 5 - Il vecchio saggio

The vecchio saggio is a man in the oasi. "Vedi anche un vecchio [o]saggio[x] che dorme all[']ombra di una palma, con la schiena appoggiata al [s]tronco[x]. E['] molto anziano e le rughe del suo volto trasudano saggezza. Deve conoscere molte cose." The description is "Il vecchio saggio sonnecchia."

Instead of giving something (called the object) to the vecchio saggio:

```
say "Il vecchio non ha preso [the object]."
```

The tronco di palma is a scenery in the oasi. The description is "[mostra-palme]A giudicare dal tronco, la palma è vecchia quasi quanto il saggio. Che abbia sempre dormito appoggiato qui?".

The tappeto afghano is a cosa-calpestable. The keyword is "tappeto". The tappeto afghano is in the oasi. "Il vecchio saggio è seduto su un [o]tappeto[x] afghano originale. Deve essere veramente

comodo a giudicare da come ci dorme sopra saporitamente." The description is "[mostra-tappeti]Un tappeto afghano originale[if the player does not wear a paio di sandali]. Salire sul tappeto deve essere un vero sollievo per i piedi[end if]."

Instead of taking a cosa-calpestable (called the portante) when the cosa-calpestable supports something (called the portato):

```
say "Prima di prendere [the portante] devi togliere [the portato]."
```

After entering the tappeto afghano:

```
say "[mostra-tappeti]Sali sul [o]tappeto[x] afghano e pensi: 'E['] molto comodo![no line break]'[if the under#tappeto contains something], ma fatto un passo senti che c[']è nascosto sotto qualcosa.[otherwise][line break][end if]"
```

After touching the tappeto afghano:

```
say "[mostra-tappeti]Tocchi il [o]tappeto[x] afghano e pensi: 'E['] veramente soffice![no line break]'[if the under#tappeto contains something], ma al tatto senti che c[']è nascosto sotto qualcosa.[otherwise][line break][end if]"
```

Before doing anything except taking inventory or looking or examining or exiting or getting off or entering when the player is on a cosa-calpestable (called the tappeto):

```
say "(prima scendi [da-prep the tappeto]);"
try silently exiting;
continue the action.
```

Before entering a cosa-calpestable (called the arrivo) when the player is on a cosa-calpestable (called the partenza):

```
if the arrivo is not on the partenza:
    say "(prima scendi [da-prep the tappeto]);"
    try silently exiting;
continue the action.
```

An underside called under#tappeto is part of the tappeto afghano. The bulk capacity of the under#tappeto is 5.

The paio di sandali is a wearable thing. The paio di sandali is in the under#tappeto. The description is "Un comodo paio di sandali." The printed name is "paio di [o]sandali[x]".

After looking under the tappeto afghano for the first time:

```
award 15 points;
continue the action.
```

Instead of going somewhere when the player does not wear the paio di sandali and the oasi is visited:

```
say "Prima di camminare, pensi: 'E['] stupido ferirsi ancora i piedi quando potrei appoggiarli su qualcosa di soffice[if the player carries the paio di sandali] o, addirittura, calzarli con questo bel paio di [o]sandali[x] che ho in mano[end if]."
```

Understand "calza" as wearing.

Instead of taking the tappeto afghano:

```
say "Non puoi prenderlo [scopri-afghano]".
```

Instead of pulling the tappeto afghano:

```
say "Non puoi tirarlo [scopri-afghano]".
```

To say scopri-afghano:

```
say "[mostra-tappeti]perché c[']è il vecchio saggio seduto sopra.[if the under#tappeto contains something]Nel tirarlo, comunque, ti accorgi di un rigonfiamento e pensi: 'Che ci sia sotto qualcosa?'[end if]".
```

After putting something (called the nascosto) on an underside (called the sottotappeto):

```
say "[mostra-tappeti]Hai nascosto [the nascosto] sotto [a-prep the sottotappeto]."
```

A puzzle is a kind of thing. A puzzle can be risolto or mescolato.

The cubo di Rubik is a puzzle. The keyword is "cubo". The cubo di Rubik is mescolato. The printed name is "[o]cubo[x] di Rubik".

The cubo di Rubik is in the under#tappeto. The description is "Il mitico cubo di Rubik! [if the cubo is mescolato] E['] mescolato e pensi: 'Chissà se sono ancora capace di risolverlo!'[otherwise] E['] già risolto e pensi: 'Chissà se sarei ancora capace di risolverlo, dopo averlo mescolato.'[end if]".

Mixing is an action applying to one thing.

Solving is an action applying to one thing.

Understand "risolvi [something]" or "risolvi [art-det] [something]" or "risolvi [a known thing]" as solving.

Understand "mescola [something]" or "mescola [art-det] [something]" or "mescola [a known thing]" as mixing.

Check mixing something (called the cubo):

```

if the cubo is not carried by the player:
    say "Non hai in mano [the cubo]." instead;
otherwise if the cubo is not a puzzle:
    say "[The cubo] non si può mescolare." instead.

```

```

Check solving something (called the cubo):
if the cubo is not carried by the player:
    say "Non hai in mano [the cubo]." instead;
otherwise if the cubo is not a puzzle:
    say "[The cubo] non si può risolvere." instead.

```

```

Carry out mixing something (called the cubo):
if the cubo is mescolato:
    say "E['] già mescolato, ma rimescolarlo ancora non fa male.";
otherwise:
    say "Mescoli [the cubo] con cura.";
now the cubo is mescolato.

```

```

Carry out solving something (called the cubo):
if the cubo is mescolato:
    now the cubo is risolto;
    say "In meno di dieci secondi risolvi [the cubo]. Nell[']ammirarlo pensi: '[one
of]E['] come cavalcare un cammello: certe cose non si dimenticano[or]Nonostante siano passati anni,
sono sempre in gamba[at random].'";
otherwise:
    say "E['] già risolto!".

```

Section 6 - Il serraglio

The serraglio is a room. It is south of the oasis. "Nel serraglio si radunano le grandi carovane di passaggio. E['] un[']area recintata in cui i mercanti possono trascorrere la notte in piena sicurezza. Oggi è deserto, ma di solito c[']è sempre una grande confusione.[line break]A [d]nord[x] torni al centro dell[']oasi, mentre a [d]ovest[x] c[']è un passaggio per raggiungere le [s]stalle[x]." [%\$] The top-illustration is figure of serraglio. [%\$]

The stalle is a female, plural-named backdrop in the serraglio. "Bassi edifici in legno usati come ricovero per le bestie da soma."

Section 7 - Il palmeto

The palmeto dell'oasi is a room. It is west of the oasis. "Le [s]palme[x] da dattero crescono una accanto all[']altra e costituiscono un formidabile riparo contro i raggi del [s]sole[x].[line break]A [d]est[x] torni al centro dell[']oasi; a [d]nord[x] si apre l[']accesso per il giardino incantato, il luogo più fresco di tutta l[']oasi; a [d]sud[x] c[']è l[']ingresso delle [s]stalle[x], dove dimorano le bestie da soma di passaggio; infine a [d]ovest[x] c[']è la sorgente d[']acqua dell'oasi, l[']unica fonte nel raggio di decine di miglia in questo deserto infuocato." [%\$] The top-illustration is figure of palmeto-dell-oasi. [%\$]

The stalle is a female, plural-named backdrop in the palmeto.
The sole is a backdrop in the palmeto.
The palme is a female, plural-named backdrop in the palmeto.

Section 8 - Il filatoio

The filatoio dell'oasi is a room. It is north of the oasis. "Il filatoio è la fonte principale di reddito per gli abitanti del villaggio. Qui le donne filano i peli di cammello producendo robusti fili di lana di cammello e tessono tappeti e indumenti, tra cui tappeti di lana di cammello e cappotti (di cammello). Ovunque vedi [s]matasse[x] di fili di lana di peli di cammello pronte per essere tessute.[line break]A [d]sud[x] torni al centro dell[']oasi, mentre ad [d]est[x] entri nel magazzino." [%\$] The top-illustration is figure of filatoio-dell-oasi. [%\$]

The matasse di fili di lana are female, plural-named scenery in the filatoio dell'oasi. The description is "[mostra-matasse]Lunghe matasse di fili di lana di cammello."

To say mostra-matasse:
[%\$] show-side figure of matasse; [%\$]
do nothing.

Instead of doing something except going somewhere in the presence of the filatrice when the player is nudo:

```

say "La filatrice continua ad emettere un grido acuto. E['] meglio che ti allontani
velocemente prima che arrivino gli uomini del villaggio e ti facciano passare per sempre la voglia
di esibirti davanti alle loro donne."

```

Instead of doing something except examining the filatrice when the player is nudo:
say "L[']urlo ti paralizza e non riesci ad agire."

The magazzino is a room. It is east of the filatoio dell'oasi. "Nel magazzino vengono conservati tutti i filati, i tessuti ed i tappeti prodotti dalle filatrici. Ad [d]ovest[x] torni nel filatoio."

[$\$$] The top-illustration is figure of magazzino. [$\$$]

The tavolo da esposizione is an enterable supporter in the magazzino. The keyword is "tavolo". The description is "[mostra-tappeti]Un robusto tavolo per esporre i tappeti."

To say mostra-tappeti:

```
[ $\$$ ] show-side figure of tappeti; [ $\$$ ]
do nothing.
```

The tappeto persiano is a cosa-calpestable on the tavolo da esposizione. The keyword is "persiano". The description is "[mostra-tappeti]Un tipico tappeto persiano dai tipici disegni, annodato tipicamente a mano. L[']etichetta del prezzo, tipica anch[']essa, recita: 'Original Al-Kawaz. 100 dollars.'[line break]Volti un angolo del tappeto per ammirarne la trama e noti una seconda etichetta, meno tipica della prima, che dice: 'Made in China'[if the under#persiano contains something].[line break]In un angolo del tappeto noti un rigonfiamento, come se ci fosse sotto qualcosa[end if]."

The tappeto volante is a cosa-calpestable on the tavolo da esposizione. The description is "[mostra-tappeti]Incredibile! Un vero tappeto volante! L[']etichetta esplicativa riporta: 'Original Flying Carpet. 50 dollars.'[line break]Volti un angolo del tappeto trovando una seconda etichetta che dice: 'Read instructions before use'[if the under#volante contains something].[line break]In un angolo del tappeto noti un rigonfiamento, come se ci fosse sotto qualcosa[end if]."

A cosa-calpestable can be pagato. A cosa-calpestable is usually not pagato.

An underside called under#persiano is part of the tappeto persiano. The bulk capacity of the under#persiano is 5.

An underside called under#volante is part of the tappeto volante. The bulk capacity of the under#volante is 5.

After entering the tappeto persiano:

```
say "[apprezza-tappeto][if the under#persiano contains something],' ma fatto un passo senti
che c[']è nascosto sotto qualcosa.[otherwise].'[line break][end if]"
```

After touching the tappeto persiano:

```
say "[apprezza-al-tatto][if the under#persiano contains something],' ma al tatto senti che
c[']è nascosto sotto qualcosa.[otherwise].'[line break][end if]"
```

After entering the tappeto volante:

```
say "[apprezza-tappeto][if the under#volante contains something],' ma fatto un passo senti
che c[']è nascosto sotto qualcosa.[otherwise].'[line break][end if]"
```

After touching the tappeto volante:

```
say "[apprezza-al-tatto][if the under#volante contains something],' ma al tatto senti che
c[']è nascosto sotto qualcosa.[otherwise].'[line break][end if]"
```

To say apprezza-tappeto:

```
say "[mostra-tappeti]Sali sul tappeto e pensi: '[one of]Questi tappeti annodati sono davvero
soffici[or]La lana di cammello è un piacere per la pianta del piede[or]Camminare sui tappeti è
veramente piacevole[at random]"
```

To say apprezza-al-tatto:

```
say "[mostra-tappeti]Tocchi il tappeto e pensi: 'Questi tappeti annodati sono veramente
soffici"
```

The taking action has an object called the place taken from.

Setting action variables for taking:

```
change the place taken from to the holder of the noun.
```

After taking the tappeto persiano when something is in the under#persiano:

```
say "[mostra-tappeti]Hai preso il tappeto persiano e scopri che sotto nascondeva [a list of
things in under#persiano].";
now everything in under#persiano is in the place taken from.
```

After taking the tappeto volante when something is in the under#volante:

```
say "[mostra-tappeti]Hai preso il tappeto volante e scopri che sotto nascondeva [a list of
things in under#volante].";
now everything in under#volante is in the place taken from.
```

Instead of putting a cosa-calpestable (called the copertura) on something (called the nascosto):

```
if the nascosto is a supporter:
    continue the action;
otherwise:
    try dropping the copertura;
    move the nascosto under the copertura.
```

The libretto di istruzioni is a thing in the under#volante. The description is "Dev[']essere il libretto d[']istruzioni del tappeto volante. Riporta lo stesso breve testo in alcune decine di

lingue tra cui l'islandese ed il coreano. [if the libretto is tradotto]Il contadino islandese lo ha tradotto così:[paragraph break][italic type]Tappeto volante per asini (solo un asino infatti potrebbe credere che un tappeto può volare)[line break]Istruzioni: mettete l[']asino sopra il tappeto e un po['] di terra del luogo di destinazione sotto il tappeto. Subito l[']asino volerà verso la meta prescelta[roman type]'[else]Sfogli con pazienza, ma rimani deluso perché, tra caratteri cinesi, runici, ebraici e cirillici, non trovi la traduzione nella tua lingua[end if]."
The libretto can be tradotto. The libretto is not tradotto.

Before of entering a cosa-calpestable (called the tappeto):
if the tappeto is carried:
say "[mostra-tappeti]Prima di salire [su-prep the tappeto] devi stenderlo." instead;
otherwise if the player wears the paio di sandali:
say "[mostra-tappeti]Prima di salire [su-prep the tappeto] devi toglierti i [o]sandali[x] altrimenti potresti squalcirlo." instead;

Understand "stendi [something]" or "stendi [art-det] [something]" as dropping.

After entering a cosa-calpestable (called the copertura):
say "[apprezza-tappeto]".

After dropping the cosa-calpestable (called the copertura):
say "Stendi [the copertura]."

After putting the cosa-calpestable (called the copertura) on something (called the coperto):
say "Stendi [the copertura] [su-prep the coperto]."

Section 9 - Il giardino incantato

The giardino incantato is a room. It is north of the palmeto dell'oasi. "Il giardino è il luogo più tranquillo e riposante dell[']intera oasi. Un vero angolo di paradiso nascosto al riparo del fitto [s]palmeto[x]. A [d]sud[x] torni al palmeto." [%\$] The top-illustration is figure of giardino-incantato. [%\$]

The palmeto di datteri is a backdrop in the giardino incantato.

A digging-tool is a kind of cosa-ingombrante.

The pala is a female digging-tool in the giardino incantato. "Una [o]pala[x] è piantata in un angolo del giardino. Sembra che non chieda altro che d[']essere utilizzata per scavare." The description is "[mostra-pala]Una robusta pala adatta per scavare grandi quantità di terra."

To say mostra-pala:
[%\$] show-side figure of pala; [%\$]
do nothing.

Digging is an action applying to nothing.

Understand "scava" as digging.

Check digging:
if the player does not carry a digging-tool,
say "Non puoi scavare questo terreno a mani nude." instead.

Report digging:
say "[mostra-pala]Hai scavato una buca [con-prep the random digging-tool carried by the player] e, dopo aver ammirato l[']opera, ti sei reso conto di quanto fosse pericolosa in quel punto, così hai impugnato nuovamente [the random digging-tool carried by the player] e l[']hai subito ricoperta. Non è certo il lavoro più utile che tu abbia portato a termine, ma la tua abilità con [the random digging-tool carried by the player] è aumentata."

The rosa purpurea is a female thing in the giardino incantato. The keyword is "rosa". "Vedi un fiore incantevole e molto profumato. Una [o]rosa[x] purpurea del deserto. Bisogna coltivarla con amore per anni, più di trent[']anni, perché possa fiorire. E['] bellissima!" The description is "[mostra-rosa]E['] un fiore di una bellezza straordinaria e dall[']intenso aroma. Dicono che il suo profumo possa mandare in estasi." The printed name is "[o]rosa[x] purpurea".

The rosa purpurea can be recisa or piantata. The rosa purpurea is piantata.

To say mostra-rosa:
[%\$] show-side figure of rosa; [%\$]
do nothing.

Understand "fiore" as rosa purpurea.

Instead of doing something except examining or smelling with the rosa purpurea in the presence of the giardiniere for the first time:
say "[mostra-rosa]Il giardiniere è molto geloso della sua rosa. Prima ti dice: 'Guardare, annusare, ma non toccare!', poi ti colpisce sulle falangi del piede con la zappa. 'Che ti serva da lezione!', aggiunge compiaciuto.[line break]Il dolore è così acuto che hai quasi l[']impressione di

uscire dal tuo stesso corpo, ma forse è solo un effetto del profumo della rosa purpurea."

Instead of doing something except examining or smelling with the rosa purpurea in the presence of the giardiniere for more than one time:

```
say "[mostra-rosa]Il giardiniere ti colpisce subito sulle falangi del piede con la zappa e ride compiaciuto. C[']è sicuramente qualcos[']altro da fare con una rosa che non procuri tutto questo dolore."
```

Instead of taking the rosa purpurea when the rosa purpurea is piantata:

```
say "Il gambo della [o]rosa[x], cresciuto per trent[']anni, è particolarmente robusto. Dovresti trovare qualcosa per reciderlo."
```

Instead of smelling the rosa purpurea for the first time:

```
award 15 points;
say "[mostra-rosa]Il profumo della rosa è inebriante, tanto che per un istante perdi coscienza di te stesso ed hai una visione...[paragraph break]
E['] proprio come se la tua anima avesse preso il volo e, contro tutte le leggi dello spazio e del tempo, avesse attraversato l[']oasi ed il deserto di sabbia e si ritrovasse, senza comprenderne il motivo, a fianco di un avvoltoio che sta volando ad est verso le montagne.[paragraph break]
L[']avvoltoio ti guarda stupito e ti dice: 'Se gli asini volassero...', ma lo interrompi e concludi: 'oscurerebbero il cielo!'.[paragraph break]
L[']avvoltoio, decisamente contrariato, ti dice: 'Questa visione ha preso una direzione sbagliata. Avresti dovuto virare a nord, sopra il deserto di sassi, o al limite a sud, sopra il deserto di pietre,...[paragraph break]
e improvvisamente la visione scompare e ti ritrovi di fronte alla [o]rosa[x]."
```

To say apriti-sesamo:

```
now the porta di pietra is stregata.
```

Instead of smelling the rosa purpurea for more than one time:

```
say "[mostra-rosa]Il profumo della rosa è inebriante, tanto che per un istante perdi coscienza di te stesso ed hai una visione...[paragraph break][one of]
E['] come se la tua anima, lasciato il corpo di fronte alla rosa, avesse attraversato l[']oasi, si fosse voltata a nord, attraverso il deserto di sassi, poi, ancora oltre, fosse giunta di fronte all[']entrata di una grotta chiusa da un[']enorme pietra.[paragraph break]
Ecco che un uomo, un gigante di grande ferocia, forse il capo della famosa banda dei quaranta ladroni, si pone di fronte alla pietra e, facendosi piccolo piccolo, bisbiglia: 'Apriti sesamo.'[apriti-sesamo]Immediatamente il terreno comincia a sussultare e vedi la pietra che chiudeva la grotta farsi da parte di colpo...
[or]
E['] come se il tuo corpo, lasciata l[']anima di fronte alla rosa, fosse trascinato da una forza invisibile attraverso l[']oasi e, voltato il cammino a sud, attraverso il deserto di pietre, fosse giunto di fronte ad un enorme alverare. Ecco che un[']ape operaia entra all[']interno e, attraverso i meandri del favo, ti conduce davanti all[']ape regina circondata da ancelle e da fuchi pigramente adagiati ai suoi piedi.[paragraph break]
L[']ape regina si rivolge a te e inorridita ti chiede: 'Come osi presentarti a me [if the player is nudo]completamente nudo[otherwise]coperto solo da un grembiule da giardiniere[end if]?'...
[in random order]
[paragraph break]ma improvvisamente la visione scompare e ti ritrovi di fronte alla [o]rosa[x]."
```

Instead of cutting the rosa purpurea:

```
now the rosa purpurea is recisa;
now the player carries the rosa purpurea;
say "[mostra-rosa]Dopo un attimo di esitazione, hai reciso la rosa purpurea."
```

The pollice verde is part of the giardiniere. The description is "E['] incredibile! Il giardiniere ha davvero il pollice verde... e io che pensavo fosse solo un modo di dire!"

The grembiule is a wearable thing. The grembiule is worn by the giardiniere. "Vedi un [o]grembiule[x] da giardiniere dotato di capienti [o]tasche[x]. Copre solo l[']essenziale, ma è molto meglio che andare in giro nudo." The description is "Un robusto grembiule da giardiniere con due grandi [o]tasche[x][if the player wears the grembiule]. Peccato che non copra il... retro[end if]."

The tasche is a female, plural-named container. It is part of the grembiule. Understand "tasca" as tasche. The description is "Queste tasche sono molto capienti. Ci può stare di tutto![if the number of things in the tasche is greater than 0] Ora contengono [list of things contained in the tasche with indefinite articles].[end if]"

The forbici is a female, plural-named thing in the tasche. The description is "[mostra-forbici]Forbici molto affilate. Con un paio di queste si può tagliare di tutto, dai gambi di rosa, ai pezzi di carne di gnu, fino alle spesse funi di peli di cammello." The printed name is "[o]forbici[x]."

To say mostra-forbici:

```
[$%] show-side figure of forbici; [%$]
do nothing.
```

Understand "forbice" as forbici.

A spruzzatore is a kind of thing. A spruzzatore can be agitato. The spruzzatore is usually not agitato.

The bomboletta spray is a female spruzzatore in the tasche. The description is "[mostra-spray]Una bomboletta spray. La didascalia stampata sull[']involucro dice: '[italic type]Repellente per api e vespe. Agita la bomboletta prima dell[']uso. Spruzza la bomboletta sugli insetti. Non disperdere nell[']ambiente.[roman type]'". The printed name is "bomboletta [o]spray[x]".

To say mostra-spray:
 [%] show-side figure of spray; [%]
 do nothing.

After dropping the bomboletta spray:
 say "[mostra-spray]Hai buttato la bomboletta [o]spray[x] per terra. Non hai proprio una coscienza civica!"

Spraying it on is an action applying to two things.

Understand "spruzza [something] [su-art] [something]" or "spruzza [art-det] [something] [su-art] [something]" as spraying it on.

Instead of waving something (called the shaker):
 if the player does not carry the shaker:
 say "Per agitare qualcosa, devi tenerlo in mano." instead;
 otherwise if the shaker is not a spruzzatore:
 say "Non serve agitare [the noun]." instead;
 otherwise:
 now the shaker is agitato;
 say "[mostra-spray]Hai agitato con cura [the shaker]."

Check spraying something (called the sprayer) on something (called the sprayed):
 if the sprayer is not a spruzzatore:
 say "Non puoi spruzzare [con-prep the sprayer]." instead;
 if the player does not carry the sprayer:
 say "Per spruzzare, devi avere in mano [the sprayer]." instead;
 otherwise if the the sprayed is not visible:
 say "Per spruzzare su qualcosa, devi vederlo." instead;
 otherwise if the sprayer is not agitato:
 say "[mostra-spray][The sprayer] non funziona. Leggi le istruzioni!" instead.

Report spraying something (called the sprayer) on something (called the sprayed):
 now the bomboletta spray is not agitato;
 say "[mostra-spray]Hai spruzzato [the sprayer] [su-prep the sprayed], ma senza alcun effetto."

Instead of inserting a cosa-ingombrante (called the oggetto) into the tasche:
 say "Non puoi. Le tasche sono capienti, ma [the oggetto] non ci sta[if the oggetto is plural-named]nno[end if]."

Instead of inserting a cosa-calpestable (called the oggetto) into the tasche:
 say "Non puoi. Le tasche sono capienti, ma [the oggetto] non ci sta."

After wearing the grembiule for the first time:
 award 50 points;
 say "[indossa-grembiule]";
 if the giardiniere is in the location:
 say "Il giardiniere si appoggia alla zappa, ti osserva e dice: 'Bravo ragazzo! C[']è sempre bisogno di nuovi apprendisti giardinieri.'"

After wearing the grembiule for more than one time:
 say "[indossa-grembiule]".

To say indossa-grembiule:
 say "E['] molto meglio che essere completamente nudi. Certo, lascia scoperto il posteriore, ma in fondo sei piuttosto orgoglioso delle tue... terga."

The zappa is a female digging-tool. The zappa is carried by the giardiniere. The description is "Una robusta zappa adatta a smuovere piccole quantità di terra."

Instead of giving the favo to the giardiniere:
 remove the giardiniere from play;
 now the zappa is in the giardino incantato;
 say "[mostra-favo]Le api operose che si aggirano nel favo prendono il volo. [fuga-giardiniere]".

To say mostra-favo:
 [%] show-side figure of favo; [%]
 do nothing.

Instead of dropping the favo when the player is in the giardino incantato and the giardiniere is on-stage:

```
remove the giardiniere from play;
now the zappa is in the giardino incantato;
say "[mostra-favo]Il favo, caduto a terra, spaventa alcune api che si alzano in volo. [fuga-giardiniere]".
```

Instead of waving the favo when the player is in the giardino incantato and the giardiniere is on-stage:

```
remove the giardiniere from play;
say "[mostra-favo]Agiti il favo ed alcune api prendono il volo. [fuga-giardiniere]".
```

To say fuga-giardiniere:

```
award 50 points;
say "[mostra-favo]Il giardiniere alla vista della api si sente soffocare e, spaventato, con una corsa precipitosa, salta la recinzione del giardino. In pochi istanti, scompare dalla vista."
```

Section 10 - Conversazione con il giardiniere

The pala is speakable. The printed name is "[o]pala[x]".
 The zappa is speakable. The printed name is "[o]zappa[x]".
 The grembiule is speakable. The printed name is "[o]grembiule[x]".
 The lavoro GRD is a discussable. The printed name is "[t]lavoro[x]".
 The vestito GRD is a discussable. The printed name is "[t]vestito[x]".
 The visioni GRD is a female, plural-named discussable. The printed name is "[t]visioni[x]".
 The fuga GRD is female discussable. The printed name is "[t]fuga[x]".
 The incantesimo GRD is discussable. The printed name is "[t]incantesimo[x]".
 The api GRD is female, plural-named discussable. The printed name is "[t]api[x]".
 The miele GRD is a discussable. The printed name is "[t]miele[x]".
 The fiori GRD is plural-named discussable. The printed name is "[t]fiori[x]".
 The lavoro GRD, vestito GRD, visioni GRD, fuga GRD, incantesimo GRD, api GRD, miele GRD, fiori GRD are part of the giardiniere.

The giardiniere is a man in the giardino incantato. "C[']è un [o]giardiniere[x] [if the giardiniere is the current interlocutor]che, appoggiato ad una zappa, sta conversando con te[otherwise]intento a zappare. Maneggia la zappa con maestria, ma è meglio tenere le debite distanze[end if][if the grembiule is off-stage]. Indossa un robusto grembiule[end if]." The description is "Il giardiniere[if the giardiniere is the current interlocutor], appoggiato alla zappa, sta conversando con te[otherwise] è intento a zappare. Se vuoi parlargli, prova prima a salutarlo con un [t]ciao[x] [end if][if pollice is unexamined]. Si capisce subito dal [o]pollice[x] verde che ama molto il suo lavoro[end if]."

giardiniere-node is a closed convnode.
 The node of the giardiniere is giardiniere-node.
 The ask-suggestions are {lavoro GRD, vestito GRD, pala}.
 The tell-suggestions are {fiori GRD, zappa}.

After saying hello to the giardiniere when the greeting type is explicit:

```
say "'Salute buon giardiniere!' esordisci.[line break]Il giardiniere interrompe il suo [t]lavoro[x] [add lavoro GRD ask suggestion], ti osserva[if the player is nudo] senza un [t]vestito[x][add vestito GRD ask suggestion] addosso[end if] e dice: 'Salute a te!'"
```

After saying hello to the giardiniere when the greeting type is implicit:

```
say "Attiri l[']attenzione con un colpo di tosse. Il giardiniere, smette di zappare, si volta e attende che tu inizi a parlare."
```

After saying goodbye to the giardiniere when the farewell type is explicit:

```
say "'Bene, allora ti saluto!' concludi.[line break]'Alla prossima,' risponde e torna al suo lavoro."
```

After saying goodbye to the giardiniere when the farewell type is implicit:

```
say "Il giardiniere ti saluta e ricomincia a zappare."
```

Default Response of giardiniere-node:

```
say "'Hai qualcosa di cui parlare o da chiedere? Altrimenti torno a lavorare,' dice il giardiniere con impazienza.[line break]";
try silently listing suggested topics.
```

Response of giardiniere-node when asked about the lavoro GRD:

```
say "'Ti occupi di questo giardino?' chiedi con cortesia.[paragraph break]L[']uomo risponde: 'Sì, nell[']oasi sono conosciuto come il giardiniere. Sono quarant[']anni che lo curo e non l[']ho mai abbandonato... o meglio, solo una volta mi è capitato di doverlo lasciare, ma è stato solo per breve tempo. E['] stata una vera [t]fuga[x]!' [add fuga GRD ask suggestion][remove lavoro GRD ask suggestion]".
```

Response of giardiniere-node when asked about the vestito GRD:

```
now the grembiule is familiar;
say "'Sono appena arrivato all[']oasi e non ho un vestito adatto. Da chi potrei averne uno in
```

prestito?' chiedi con urgenza. [paragraph break]'Le filatrici tessono anche vestiti[if the player is nudo], ma se ti presenti in questo modo ti caceranno sicuramente[end if]. Io [if the grembiule is worn by the giardiniere]ti posso dare solo il mio [t]grembiule[x] da giardiniere[add grembiule ask suggestion][otherwise]non ho altro da darti oltre a quanto ti ho già prestato[end if],' risponde. [remove vestito GRD ask suggestion]".

Response of giardiniere-node when asked about the grembiule:

now the grembiule is in the giardino incantato;
say "'In effetti il tuo grembiule mi farebbe davvero comodo,' e lo chiedi implicitamente con lo sguardo.[paragraph break]Il giardiniere si slaccia il [o]grembiule[x] e lo fa scivolare ai tuoi piedi dicendo: 'Prendilo pure, io ne ho altri.'[remove grembiule ask suggestion]".

Response of giardiniere-node when asked about the pala:

say "[mostra-pala]'[if the pala is in the location]Questa [o]pala[x][otherwise]La [o]pala[x] che era qui[end if] ti serve?' chiedi al giardiniere con noncuranza. [paragraph break]'La uso quando devo scavare, ma se ne hai bisogno te la posso prestare,' risponde.[remove pala ask suggestion]".

Response of giardiniere-node when told about fiori GRD:

say "[mostra-rosa]'Hai davvero dei fiori molto belli qui, giardiniere,' affermi con tono ammirato. [paragraph break]'Sì, sono la mia passione. Per questa [o]rosa[x] purpurea, ad esempio, ci sono voluti trent[']anni perché fiorisse, ma ne è valsa la pena,' risponde, 'molti poi credono che questi fiori siano sottoposti ad un [t]incantesimo[x].' [remove GRD fiori tell suggestion][add incantesimo GRD ask suggestion]".

Response of giardiniere-node when told about the zappa:

say "Con tono ammirato dici al giardiniere: 'Riesci a maneggiare quella zappa con estrema maestria, sei veramente molto esperto.' [paragraph break]Con malcelato orgoglio ti risponde: 'Ci vogliono anni d[']esercizio.' [remove zappa tell suggestion]".

Response of giardiniere-node when asked about incantesimo GRD:

say "'Un incantesimo? Di che si tratta?' chiedi con curiosità. [paragraph break]Il volto del giardiniere si vela di tristezza e ti dice: 'Molti sostengono che i profumi dei fiori di questo giardino inducano [t]visioni[x] di straordinaria bellezza.' [add visioni GRD ask suggestion][remove incantesimo GRD ask suggestion]".

Response of giardiniere-node when asked about visioni GRD:

say "Sopreso da come il volto del giardiniere si è rabbuiato e incuriosito dalle visioni, insisti: 'Visioni? Che tipo di visioni?' [paragraph break]Il volto del giardiniere si fa ancora più triste, poi ti racconta: 'Molti hanno annusato i fiori del giardino e hanno provato una sorta di proiezione astrale. La loro anima ha fluttuato al di fuori del proprio corpo. Non ti so dire altro perché, purtroppo, sono affetto da sinusite cronica e non percepisco gli odori.' [paragraph break]A questo punto ti viene in mente una simpatica barzelletta e dici: 'Ehi! Lo sai qual è il colmo per un giardiniere?' [paragraph break]Il giardiniere ti risponde seccamente: 'Lo so!' [remove visioni GRD ask suggestion]".

Response of giardiniere-node when asked about fuga GRD:

say "'Una vera fuga?' chiedi sorpreso, 'che cosa è successo?' [paragraph break]Il volto del giardiniere si fa paonazzo, poi ti racconta: 'Devi sapere che io sono allergico alle punture delle [t]api[x]. Non so perché, ma appena un[']ape mi punge, nella zona interessata cresce subito un pomfo dolorante e nel contempo la respirazione si fa difficile ed il cuore palpita. Appena vedo un[']ape mi prende l[']ansia, ma quando sono isolate riesco a tenerle sotto controllo.' [paragraph break]A questo punto dici simpaticamente: 'Ehi! mi è venuta in mente una bellissima barzelletta: ci sono un[']ape e un giardiniere...' [paragraph break]Il giardiniere ti interrompe seccamente: 'La so!' [remove fuga GRD ask suggestion][add api GRD ask suggestion]".

Response of giardiniere-node when asked about api GRD:

say "'Non sapevo che le api fossero così pericolose,' dici sorpreso, 'ce ne sono molte da queste parti?' [paragraph break]Il giardiniere ti spiega: 'Le api non sono pericolose per tutti, ma solo per chi è allergico al loro veleno, come me. Preferisco non pensare a dove sono radunate a migliaia, intorno al [t]miele[x] degli alveari... la sola idea mi rende ansioso. Comunque porto sempre con me uno spray repellente contro api e vespe per difendermi nelle situazioni di emergenza.' [remove api GRD ask suggestion][add miele GRD ask suggestion]".

Response of giardiniere-node when asked about miele GRD:

say "Dissimulando un eccessivo interesse, chiedi: 'Sai dove posso trovare del miele in questa regione? Ho sentito dire che il miele d[']acacia è molto buono.' [paragraph break]'Verso sud, oltre il deserto di pietre ci sono molti apicoltori e puoi trovare tutto il miele che vuoi,' ti spiega il giardiniere.[remove miele GRD ask suggestion]".

Section 11 - La stalla

The stalla is a female room. It is south of the palmeto dell'oasi. It is west of the serraglio.
"Nelle stalle, durante la notte, trovano rifugio le bestie da soma delle carovane.[line break]A [d]nord[x] c[']è l[']ingresso dal lato del [s]palmeto[x], mentre a [d]est[x] l[']accesso dalla parte del [s]serraglio[x]." [%\$] The top-illustration is figure of stalla. [%\$]

the palmeto di datteri is a backdrop in the stalla.

the recinto del serraglio is a backdrop in the stalla.

A tied-animal is a kind of rideable animal. A tied-animal can be tied or untied. A tied-animal is usually untied.

The asino is a tied-animal in the stalla. "[if the asino is tied]Ti segue[otherwise]Vedi[end if] un [o]asino[x] dall[']aspetto macilento[if the forziere is on the dorso], con un forziere sul dorso[end if][if the asino is untied and the orda di predoni is in the location of the asino], che cerca di sfuggire all[']orda di predoni[end if]." The description is "[mostra-asino]Lo sguardo non è molto intelligente, ma sembra buono[if the number of things on the dorso is not zero]. Porta [list of things on the dorso with indefinite articles] sul dorso[end if]."

To say mostra-asino:
 [%] play sound of asino; [%]
 [%] show-side figure of asino; [%]
 do nothing.

The dorso is a supporter. It is part of the asino.
 Understand "schiena" as dorso.

Instead of putting something (called the carico) on the asino:
 try putting the carico on the dorso.

Instead of pushing the asino:
 say "[mostra-asino]Spingi l[']asino[push-pull the asino]";

Instead of pulling the asino:
 say "[mostra-asino]Tiri l[']asino[push-pull the asino]";

To say push-pull the asino:
 if the asino is on the tappeto volante:
 now the asino is in the location;
 say " che scende dal tappeto volante.";
 otherwise if the tappeto volante is in the location:
 now the asino is on the tappeto volante;
 say " che sale sul tappeto volante.[asino-in-decollo]";
 otherwise:
 say ", ma questo fa solo un passo e poi si ferma."

After inserting the manciata di terra into the under#volante:
 say "Hai messo la manciata di terra sotto il tappeto volante.[asino-in-decollo]"

To say asino-in-decollo:
 if the asino is on the tappeto volante and the manciata di terra is in the under#volante:
 award 75 points;
 remove manciata di terra from play;
 now the tappeto volante is in the accesso alla grotta;
 say "Improvvisamente il tappeto si stacca da terra con sopra l[']asino stupefatto.

Dopo qualche attimo di disorientamento, il tappeto prende quota, si dirige verso est e l[']asino volante incontra l[']avvoltoio. Quest[']ultimo, ormai più amareggiato che sorpreso, prende in seria considerazione la possibilità di migrare verso qualche altra oasi dai cieli meno affollati. Poi, il tappeto, con l[']asino al seguito, vira a nord in direzione della grotta dei quaranta ladroni."

Instead of going somewhere when the player is carried by the asino:
 say "[mostra-asino][one of]L[']asino non si ribella ma non ha assolutamente intenzione di muoversi[or]Non ci sono bastoni o carote che tengano: l[']asino non si muove[or]L[']asino testardamente rimane fermo[at random]. Probabilmente è una bestia da soma abituata ad avere un carico molto pesante sul dorso e ad essere condotta legata ad altre bestie."

Instead of mounting the asino when the asino is affamato:
 say "[mostra-asino]L[']asino si imbezzarrisce e ti allontana non alcuni calci ben assestati.[asino-affamato]".

Instead of putting something on the asino when the asino is affamato:
 say "[mostra-asino]L[']asino rifiuta la soma, comincia a sgroppare e si allontana. [asino-affamato]".

To say asino-affamato:
 say "Sembra affamato. Dovresti provare ad offrirci qualcosa per ammansirlo."

Instead of giving something (called the zuccherino) to the asino:
 if the zuccherino is not the rosa:
 say "L[']asino non presta attenzione [a-prep the zuccherino]. C[']è sicuramente qualcosa che può saziarlo, ma non dev[']essere facile capire di cosa si tratta.";
 otherwise if the porta di pietra is not stregata:
 say "[mostra-rosa]Prima di sacrificare una [o]rosa[x] così bella, non pensi che sia meglio annusarla ancora una volta?";
 otherwise:
 now the asino is not affamato;
 award 125 points;
 remove the rosa from play;

say "[mostra-asino]L[']asino invece di annusare la rosa, ne fa un solo boccone. Ne mastica i petali lentamente con aria estasiata e, dopo averla inghiottita, raglia riconoscente."

Section 12 - La sorgente dell'oasi

The sorgente dell'oasi is a female room. It is west of the palmeto dell'oasi. "Ecco la [s]fonte[x] inesauribile che permette all[']intera oasi di sopravvivere all[']interno di un deserto infuocato. La fonte d[']acqua scaturisce da una [s]roccia[x] e si getta in un [s]laghetto[x]. Si pensa che l[']acqua che qui sgorga sia quella raccolta grazie alle precipitazioni sulle montagne che si trovano ad oriente.[line break]Ad [d]est[x] torni nel [s]palmeto[x]." [%\$] The top-illustration is figure of sorgente-dell-oasi. [%\$]

The palmeto di datteri is a backdrop in the sorgente dell'oasi.

To say mostra-cascata:
 [%\$] show-side figure of cascata; [%\$]
 [%\$] play sound of onde; [%\$]
 do nothing.

To say tuffo-cascata:
 [%\$] show-side figure of cascata; [%\$]
 [%\$] play sound of tuffo; [%\$]
 do nothing.

The otri pieni is a plural-named cosa-ingombrante. The keyword is "otri". The description is "[mostra-otri]Due otri in pelle d[']asino ricolmi d[']acqua." The printed name is "[o]otri[x] pieni".

To say mostra-otri:
 [%\$] show-side figure of otri; [%\$]
 do nothing.

The roccia della fonte di acqua is a female scenery in the sorgente dell'oasi. The description is "[mostra-cascata]Un potente getto d[']acqua chiara, fresca e pulita."

The laghetto is an enterable scenery container in the sorgente dell'oasi. The description is "[mostra-cascata]Un bel laghetto poco profondo. Invoglia a fare il bagno."

Instead of taking the laghetto:
 say "Prendi l[']acqua nel cavo della mano e ti disseti."

Instead of inserting something (called the object) into the laghetto:
 say "Non getti [the object] nel laghetto perché non vuoi inquinare l[']acqua."

Instead of entering the laghetto when the player wears the sandali:
 say "Indossi i [o]sandali[x]. Prima dovresti toglierli."

After entering the laghetto:
 say "[tuffo-cascata]Scendi nel laghetto per rinfrescarti le idee e ci entri fino al ginocchio. E['] davvero una bellissima sensazione."

After exiting when the player is on the laghetto:
 say "Esci dal laghetto con le idee più chiare, pronto per affrontare nuove avventure."

Instead of drinking the laghetto:
 say "Meglio bere direttamente alla fonte."

Instead of drinking the roccia della fonte di acqua:
 say "[mostra-cascata]Ti disseti abbondantemente alla fonte. L[']acqua è fresca e ti senti rinascere."

Instead of drinking the otri pieni:
 say "[mostra-otri]Bevi un po['] d[']acqua dagli [o]otri[x] e, anche se è calda, ti senti rinfancato."

Section 13 - Conversazione con l'acquaiolo

The otri pieni is speakable.

The roccia della fonte di acqua is speakable. The printed name is "[t]acqua[x]".

The lavoro ACQ is a discussable. The printed name is "[t]lavoro[x]".

The lavoro ACQ is part of the acquaiolo.

The otri pieni is carried by the acquaiolo.

The acquaiolo is a man in the sorgente dell'oasi. "C[']è un [o]acquaiolo[x] [if the acquaiolo is the current interlocutor]che sta conversando con te[otherwise]intento a pompare l[']acqua del [s]laghetto[x][end if]." The description is "[mostra-cascata]L[']acquaiolo [if the acquaiolo is the current interlocutor]sta conversando con te[otherwise]è intento a pompare l[']acqua. Se vuoi parlargli, prova prima a salutarlo con un [t]ciao[x][end if]."

acquaiolo-node is a closed convnode.
 The node of the acquaiolo is acquaiolo-node.
 The ask-suggestions are {lavoro ACQ}.
 The tell-suggestions are {roccia della fonte di acqua}.

After saying hello to the acquaiolo when the greeting type is explicit:
 say "[mostra-cascata]'Salute, acquaiolo!' esordisci.[line break]L'acquaiolo interrompe il suo [t]lavoro[x][add lavoro ACQ ask suggestion] e l['][t]acqua[x][add roccia della fonte di acqua tell suggestion] cessa di scorrere, ti osserva e dice: 'Salute a te!'"

After saying hello to the acquaiolo when the greeting type is implicit:
 say "[mostra-cascata]Attrai l[']attenzione con un colpo di tosse. L[']acquaiolo si alza e attende che tu inizi a parlare."

After saying goodbye to the acquaiolo when the farewell type is explicit:
 say "'Bene, allora ti saluto!' concludi.[line break]'Alla prossima,' risponde e torna a pompare."

After saying goodbye to the acquaiolo when the farewell type is implicit:
 say "L[']acquaiolo ti saluta e ricomincia a pompare."

Default Response of acquaiolo-node:
 say "'Hai qualcosa di cui parlare o da chiedere? Altrimenti torno a lavorare,' dice l[']acquaiolo con impazienza.[line break]";
 try silently listing suggested topics.

Response of acquaiolo-node when asked about the lavoro ACQ:
 say "'Gestisci l[']acqua dell[']oasi?' chiedi interessato.[paragraph break]'Sì,' risponde, 'sono l[']acquaiolo[if the acquaiolo carries the otri pieni]. Se hai sete ed hai bisogno di [t]acqua[x] per bere, puoi chiedere a me[add roccia della fonte di acqua ask suggestion][end if].'[remove lavoro ACQ ask suggestion]"

Response of acquaiolo-node when asked about the roccia della fonte di acqua:
 say "'Ho giusto la gola secca. Quanto costa un bicchiere d[']acqua?' chiedi con qualche apprensione.[paragraph break]L[']acquaiolo ti risponde: 'L[']acqua non costa niente, ne puoi bere quanta ne vuoi. Gli [t]otri[x] pieni invece costano molto: se vuoi portare via dell[']acqua devi pagare, ma non mi sembra che tu te lo possa permettere.'[remove roccia della fonte di acqua ask suggestion][add otri pieni ask suggestion]"

Response of acquaiolo-node when asked about the otri pieni:
 say "[mostra-otri]'Cosa vuoi per due otri d[']acqua da portare via?' chiedi con curiosità.[paragraph break]'Se chiedi tanto per chiedere, non meriti una risposta, ma se sei davvero interessato, la risposta non ti piacerà. Per venirti incontro potrei pensare ad un pagamento in natura...'[paragraph break]Il discorso comincia a preoccuparti, ma l[']acquaiolo continua: 'io sono sempre in mezzo all[']acqua e non bevo altro, ma sogno continuamente di assaggiare del buon vino, un novello sincero. Se tu mi portassi una botticella di vino, ti darei sicuramente due otri colmi d[']acqua.'[paragraph break]La richiesta non ti sembra eccessiva, ma cominci a chiederti dove potresti trovare del vino in un deserto.[remove otri pieni ask suggestion]"

Response of acquaiolo-node when told about the roccia della fonte di acqua:
 say "'L'acqua di questa sorgente è molto chiara,' affermi con ammirazione.[paragraph break]L[']acquaiolo risponde con orgoglio: 'Certamente! E['] la migliore [t]acqua[x] che puoi trovare nel raggio di molte miglia... ma ripensandoci, detto così, non è un gran complimento.'[add roccia della fonte di acqua ask suggestion][remove roccia della fonte di acqua tell suggestion]"

Instead of giving the botticella di vino to the acquaiolo:
 award 25 points;
 remove the botticella di vino from play;
 now the player carries the otri pieni;
 say "[mostra-otri]Dai la botticella di vino all[']acquaiolo che l[']accetta con entusiasmo. In cambio ti dà i due [o]otri[x] d[']acqua promessi. Cedere del buon vino in cambio di acqua non ti sembra equo, ma spesso nel deserto la gerarchia dei valori viene rovesciata."

Section 14 - Conversazione con la filatrice

The filatrice is a female woman in the filatoio dell'oasi. "C[']è una [o]filatrice[x] intenta nel suo lavoro.[if the player is nudo] Appena ti vede, così nudo come un verme, si mette le mani sugli occhi e comincia ad emettere un fastidioso grido acuto. E['] meglio che ti allontani velocemente! [end if]". The description is "La filatrice indossa i tipici costumi delle donne della regione ed un velo le copre naso e bocca lasciando visibili solo gli occhi. [if the filatrice is the current interlocutor]Mentre parla il velo segue con dolcezza il movimento del mento[otherwise if the player is nudo]Scioccata dalle tue nudità, sta emettendo un fastidiosissimo grido acuto[otherwise]E['] impegnata a tessere con rapidità e perizia. Se vuoi parlarle, prova prima a salutarla con un [t]ciao[x][end if]."

The matasse di fili di lana is speakable. The printed name is "[t]matasse[x]".
 The lavoro FLT is a discussable. The printed name is "[t]lavoro[x]".
 The filati FLT is a plural-named discussable. The printed name is "[t]filati[x]".
 The tappeti FLT is plural-named discussable. The printed name is "[t]tappeti[x]".

The vestiti FLT is a plural-named discussable. The printed name is "[t]vestiti[x]".
 The stoffe FLT is a female, plural-named discussable. The printed name is "[t]stoffe[x]".
 The lavoro FLT, filati FLT, tappeti FLT, vestiti FLT, stoffe FLT, filati FLT is part of the
 filatrice.

filatrice-node is a closed convnode.
 The node of the filatrice is filatrice-node.
 The ask-suggestions are {lavoro FLT}.
 The tell-suggestions are {filati FLT}.

After saying hello to the filatrice when the greeting type is explicit:
 say "'Salute, filatrice!' esordisci.[line break]La filatrice interrompe il suo [t]lavoro[x][add
 lavoro FLT ask suggestion], ti osserva e dice: 'Salute a te!'"

After saying hello to the filatrice when the greeting type is implicit:
 say "Attiri l[']attenzione con un colpo di tosse. La filatrice smette di tessere e attende che tu
 inizi a parlare."

After saying goodbye to the filatrice when the farewell type is explicit:
 say "'Bene, allora ti saluto!' concludi.[line break]'Alla prossima,' risponde e torna a tessere."

After saying goodbye to the filatrice when the farewell type is implicit:
 say "La filatrice ti saluta e ricomincia a tessere."

Default Response of filatrice-node:
 say "'Io avrei tante cose di cui parlare, ma se tu non hai nulla da chiedere, forse farei meglio a
 tornare al lavoro,' dice la filatrice con tristezza.[line break]";
 try silently listing suggested topics.

Response of filatrice-node when told about filati FLT:
 say "[mostra-filati]'Questi fili di lana che producite in matasse sembrano davvero robusti,' affermi
 con convinzione.[paragraph break]La filatrice risponde: 'Dipende da cosa ne devi fare. I tessuti che
 si possono tessere con i fili di lana di cammello sono molto resistenti, ma se vuoi ottenerne delle
 funi davvero robuste li devi intrecciare.'[remove filati tell suggestion]".

To say mostra-filati:
 [%] show-side figure of filati; [%]
 do nothing.

Response of filatrice-node when asked about lavoro FLT:
 say "Tanto per attaccare discorso, chiedi alla filatrice: 'Di cosa ti occupi?'[paragraph break]La
 filatrice, che da tutta la mattina aspettava qualcuno con cui chiacchierare, comincia ad elencare
 tutte le sue mansioni: 'Qui faccio un po['] di tutto,' dice, 'annodo [t]tappeti[x], creo
 [t]stoffe[x], cucio [t]vestiti[x], ma la mia specialità è quella di filare robusti fili di lana di
 cammello, come le [t]matasse[x] che vedi qui appese. Tanto per farti un esempio, quel grembiule da
 giardiniere che indossi con disinvoltura è stato prodotto qui e, come avrai notato, è soffice al
 tatto, ma allo stesso tempo molto resistente. Se mi chiedessi come viene confezionato, ti potrei
 dire...' , ma interrompi quel fiume inarrestabile di parole giusto in tempo con un 'no, grazie, mi
 interessa relativamente' prima di venire travolto dai flutti ed affogare senza più riuscire a
 riemergere.[remove lavoro ask suggestion][add tappeti ask suggestion][add stoffe ask suggestion][add
 vestiti ask suggestion][add matasse di fili di lana ask suggestion]".

Response of filatrice-node when asked about tappeti FLT:
 say "[mostra-tappeti]'Cosa mi puoi dire sui tappeti che annodi?' chiedi con circospezione.[paragraph
 break]'Annodo tappeti di ogni tipo: persiani, afgani, pakistani. Purtroppo confezionarli richiede
 molto tempo e per contenere le spese vendiamo, ai turisti di bocca buona, anche tappeti importati di
 poco prezzo, ma nel tuo caso si vede che te ne intendi e non ti lascerai sicuramente trarre in
 inganno. Nel magazzino ne abbiamo ancora qualcuno. C[']è anche un interessante tappeto volante in
 offerta. Se mi chiedessi come viene prodotto un tappeto volante, ti potrei raccontare...' , ma la
 interrompi appena in tempo con un 'no, grazie, la confezione dei tappeti volanti non è il mio
 argomento preferito', quando ormai l[']onda di piena del discorso ti aveva quasi travolto.[remove
 tappeti ask suggestion]".

Response of filatrice-node when asked about stoffe FLT:
 say "'Dimmi invece delle stoffe che produci,' le chiedi con cautela.[paragraph break]'Non mi
 crederai, ma le stoffe di questo filatoio sono note in tutta la regione...' [paragraph break]'Non
 stento a crederlo,' pensi tra te e te, 'c[']è la fila qui fuori di cammelli che vogliono farsi un
 vestito.' [paragraph break]'...eppure non facciamo nulla per pubblicizzarle,' continua, 'infatti, non
 ne abbiamo bisogno, è materiale che si vende da solo. Osserva come si sovrappongono la trama e
 l[']ordito: è un tessuto compatto adatto per ogni uso. Gli abiti confezionati con questo tessuto
 durano una vita. Se tu volessi sapere quanti vestiti hanno prodotto questi telai...' , ma la zittisci
 prontamente con un 'no, grazie, penso che riuscirò a dormire lo stesso stanotte', mentre l[']onda
 oceanica che stavi cavalcando sulla tavola con grande perizia ti aveva quasi disarcionato.[remove
 stoffe ask suggestion]".

Response of filatrice-node when asked about vestiti FLT:
 say "'In effetti, io avrei bisogno di un abito,' dici guardando il grembiule che ti copre a fatica.
 [paragraph break]'Purtroppo abbiamo spedito l[']ultimo stock di abiti con la carovana partita questa
 notte. C[']è molta richiesta in questo periodo ed il magazzino è rimasto sfornito, ma se hai un

po['] di pazienza in una settimana riusciremo ad evadere le prenotazioni dei grossisti e potremo accontentare i clienti occasionali. Tu non immagini quale sia il modello che va più in voga al momento...', ma la interrompi con decisione con un 'no, non lo immagino proprio', mentre il tornado di parole stava quasi per acquistare forza.[remove vestiti ask suggestion]".

Response of filatrice-node when asked about matasse di fili di lana:
 now the matassa di fili di lana is carried by the player;
 say "[mostra-matasse]'Vedo qui molte matasse di fili di lana di peli di cammello e sarei curioso di saggiarne la resistenza. Se ne può avere un campione?' chiedi con noncuranza.[paragraph break]'Prendi pure questa matassa, saggiane la qualità e non resterai deluso. Ricorda che le vendiamo con sconti del 30 per cento a partire da carichi da dieci cammelli. Non puoi immaginare quanto sia richiesta questa lana per fare...', ma interrompi il discorso, evitando di suggerire: 'cappotti di cammello?', e ringrazi per la [o]matassa[x] di fili di lana.[remove matasse di fili di lana ask suggestion]".

The matassa di fili di lana is a female thing. The keyword is "matassa". The description is "[mostra-filati]Una matassa di robusti fili di lana di cammello." The printed name is "[o]matassa[x]".

The fune is a female thing. The description is "[mostra-fune]Una robusta fune di peli di cammello." The printed name is "[o]fune[x]".

To say mostra-fune:
 [%] show-side figure of fune; [%]
 do nothing.

Understand "lega [something] e [something]" as tying it to.

Instead of tying something (called the first) to something (called the second):
 if the player does not carry the fune and the player does not carry the matassa di fili di lana:
 say "Non hai nulla in mano con cui legare [the first] [a-prepare the second].";
 otherwise if the first is not in the location and the second is not in the location:
 say "[The first] e [the second] non si possono legare se non li hai entrambi davanti.";
 otherwise if the player carries the matassa di fili di lana:
 say "[mostra-filati]Provi a legare [the first] [a-prepare the second] ma i fili di peli di cammello non sono abbastanza resistenti e si spezzano. Ci vuole qualcosa di più robusto.";
 otherwise if (the first is the asino and the second is the cammello) or (the first is the cammello and the second is the asino):
 now the asino is tied;
 now the fune is on the dorso;
 say "[mostra-fune]Hai legato [the first] [a-prepare the second].";
 otherwise:
 say "Non serve legare [the first] [a-prepare the second]."

Before cutting something (called the rope) when the player does not carry the forbici:
 say "Non hai in mano nessun attrezzo per tagliare [the rope]." instead.

Instead of cutting something (called the rope):
 say "Non serve tagliare [the rope].";

Instead of cutting the fune:
 if the fune is on the dorso:
 now the player carries the fune;
 now the asino is untied;
 say "[mostra-fune]Hai tagliato la [o]fune[x] e ora l[']asino è libero.";
 otherwise:
 say "La fune non è legata. E[']inutile tagliarla."

Understand "intreccia [something]" or "intreccia [art-det] [something]" as turning.

Instead of turning the matassa di fili di lana:
 award 75 points;
 remove matassa di fili di lana from play;
 now the fune is carried by the player;
 say "[mostra-fune]Intrecci i fili della matassa numerose volte fino ad ottenere una robusta [o]fune[x]."

Section 15 - Conversazione con il magazziniere

The magazziniere is a man in the magazzino. "C[']è un [o]magazziniere[x] alla [s]cassa[x] [if the magazziniere is the current interlocutor]che sta parlando con te[otherwise]intento a contare il [s]denaro[x] incassato[end if]." The description is "Un uomo molto pignolo e robusto: un vero magazziniere. E['] seduto dietro alla [s]cassa[x] e [if the magazziniere is the current interlocutor]sta parlando con te[otherwise]sta contando il [s]denaro[x] incassato. Se vuoi parlargli, prova prima a salutarlo con un [t]ciao[x][end if]."

The cassa is a female scenery in the magazzino. The description is "Una cassa vecchio modello, di

quelle che si aprono con la manovella."

The denaro incassato is scenery in the magazzino. The description is "Deve essere il denaro incassato nell[']ultima compravendita. Il magazziniere lo sta contando con rapacità, meglio stare alla larga."

The tappeto volante is speakable. The printed name is "tappeto [o]volante[x]".
 The libretto di istruzioni is speakable. The printed name is "[o]libretto[x] di istruzioni".
 The tappeti MGZ is a plural-named discussable. The printed name is "[t]tappeti[x]".
 The assortimento MGZ is a discussable. The printed name is "[t]assortimento[x]".
 The pagamenti MGZ is a plural-named discussable. The printed name is "[t]pagamenti[x]".

The tappeti MGZ, assortimento MGZ, pagamenti MGZ are part of the magazziniere.

magazziniere-node is a closed convnode.
 The node of the magazziniere is magazziniere-node.
 The ask-suggestions are {assortimento MGZ}.
 The tell-suggestions are {tappeti MGZ}.

After saying hello to the magazziniere when the greeting type is explicit:
 say "'Salute a te!' esordisci.[line break]Il magazziniere risponde: 'Salute!'"

After saying hello to the magazziniere when the greeting type is implicit:
 say "Attiri l[']attenzione con un colpo di tosse. Il magazziniere ti nota e smette di contare il denaro."

After saying goodbye to the magazziniere when the farewell type is explicit:
 say "'Addio!' concludi.[line break]'Addio!' risponde il magazziniere e ricomincia a contare il denaro."

After saying goodbye to the magazziniere when the farewell type is implicit:
 say "Il magazziniere riprende a contare il denaro."

Default Response of magazziniere-node:
 say "'Hai qualcosa di cui parlare o da chiedere?' dice il magazziniere con impazienza.[line break]";
 try silently listing suggested topics.

Response of magazziniere-node when asked about the assortimento MGZ:
 say "[mostra-tappeti]'Questo magazzino non è molto ben fornito, come mai?' chiedi con il classico tono del cliente deluso... e rompicatole.[paragraph break]Il magazziniere ti spiega pazientemente: 'La carovana partita questa notte ha comprato anche i fondi di magazzino. Sono rimasti solo due tappeti: un persiano e un tappeto [o]volante[x]'.[add tappeto volante ask suggestion][remove assortimento MGZ ask suggestion]".

Response of magazziniere-node when asked about tappeto volante:
 say "[mostra-tappeti]'Ho notato il tappeto volante. Come funziona?' chiedi con curiosità.[paragraph break]Il magazziniere ti risponde: 'Non lo so, ma da qualche parte ci dev[']essere il [o]libretto[x] di istruzioni'.[add libretto di istruzioni ask suggestion][remove tappeto volante ask suggestion]".

Response of magazziniere-node when asked about libretto di istruzioni:
 say "'Le istruzioni per il tappeto volante sono scritte in decine di lingue, alcune anche morte, ma io non ne conosco nessuna. Come faccio?' chiedi spazientito.[paragraph break]Il magazziniere replica imperturbabile: 'Hai pensato di seguire un corso di lingue?'[paragraph break]Abbozzi e non replichi.[remove libretto di istruzioni ask suggestion]".

Response of magazziniere-node when asked about pagamenti MGZ:
 say "Un po['] preoccupato chiedi: 'Cosa intendi per... pagamenti in natura?'[paragraph break]Il magazziniere risponde candidamente: 'Tu porti qualcosa che abbia un valore paragonabile al bene che vuoi comprare, lo consegni e ti porti via il bene... di cosa pensavi che si trattasse?'[paragraph break]Imbarazzato dici: 'Ehm, non so, non avevo idea... comunque, che cosa può valere quanto i tappeti che sono qui?'[paragraph break]Il magazziniere ci pensa un attimo, fa due conti, poi ti risponde: 'Vediamo... il tappeto volante costa 50 dollari, mentre il persiano il doppio. Potresti trovare qualcosa di raro e abbastanza prezioso, che so... la prima cosa che mi viene in mente sono le grosse uova d[']avvoltoio, che sono molto rinomate dai ristoratori del luogo.'[line break]Il magazziniere, meditando tra sé e sé sulla prelibatezza delle grosse uova d[']avvoltoio, continua: 'Se ne porti una ti puoi prendere il tappeto volante, con due ti prendi invece il persiano e con tre - mi voglio rovinare - te li aggiudichi entrambi. Ma ti potrei fare altri esempi di pagamenti in natura!'[paragraph break]'No,' lo interrompi prontamente, 'questo è già abbastanza chiaro!'[remove pagamenti MGZ ask suggestion]".

Response of magazziniere-node when told about tappeti MGZ:
 say "[mostra-tappeti]'Questi tappeti sono molto belli: se avessi dei soldi ne comprerei sicuramente uno,' dici tastando il terreno.[paragraph break]'Non tutto si può comprare con i contanti,' risponde il magazziniere, 'ci sono anche le carte di credito,' aggiunge ridendo, 'ed i [t]pagamenti[x] in natura!'[add pagamenti MGZ ask suggestion][remove tappeti MGZ tell suggestion]".

Instead of putting a cosa-calpestable (called the rubato) on the gobba when the magazziniere is visible:

```
say mostra-tappeti;
if the rubato is pagato:
```

```

    say "'Hai fatto un buon acquisto,' ti dice il magazziniere, 'non te ne pentirai.'";
    now the rubato is on the gobba;
otherwise:
    say "Stai tentando di caricare [the rubato] sul cammello, ma il magazziniere ti ferma
dicendo: 'Ehi! Prima pagare, poi caricare!'".

```

```

Instead of giving the uovo to the magazziniere:
    award 75 points;
    now the tappeto volante is pagato;
    remove the uovo from play;
    say "Il magazziniere prende l[']uovo e dice: 'Un grosso e prelibato uovo d[']avvoltoio! Te
lo posso valutare 50 dollari. Prendi pure quello che vuoi nel magazzino, senza però superare la
cifra,' e indica con un ampio gesto il magazzino quasi vuoto.'"

```

```

Instead of going somewhere when the player carries a not pagato cosa-calpestable (called the
rubato):
    say "Stai tentando di andartene, ma il magazziniere ti ferma dicendo: 'Ehi! Pensi di
andartene senza pagare [the rubato]?'".

```

Chapter 3 - Il deserto di sassi

The deserto di sassi is a luogo-lontano. It is north of the deserto di sabbia. The description is "Intorno a te riesci a vedere solo [s]sassi[x] arroventati dal [s]sole[x], mentre a [d]sud[x] scorgi in lontananza le [s]dune[x] del deserto di sabbia[if player is not carried by something], ma sono troppo lontane per arrivarci a piedi[end if]. A [d]nord[x] invece il deserto si trasforma nell[']inferno di sassi." [%] The top-illustration is figure of deserto-di-sassi. [%\$]

The distesa di sassi is a backdrop in the deserto di sassi.
The cielo is a backdrop in the deserto di sassi.
The sole is a backdrop in the deserto di sassi.
The dune di sabbia is a backdrop in the deserto di sassi.

The inferno di sassi is a luogo-molto-lontano. It is north of the deserto di sassi. The description is "Un[']infinità di [s]sassi[x] roventi ti circonda. Dirigendoti a [d]sud[x] puoi tornare verso il più mite deserto di sassi. A [d]nord[x] invece vedi le [s]montagne[x] magiche e, scrutando con attenzione attraverso l[']aria tremolante, noti in una [s]spaccatura[x] della roccia [if porta di pietra is stregata]l[']entrata della grotta della visione suscitata dal profumo della rosa[otherwise]quello che potrebbe essere l[']accesso ad una grotta[end if]." [%] The top-illustration is figure of inferno-di-sassi. [%\$]

The distesa di sassi is a backdrop in the inferno di sassi.
The cielo is a backdrop in the inferno di sassi.
The montagne magiche is a backdrop in the inferno di sassi. The description is "Montagne azzurre all[']orizzonte. Circondano l[']inferno di sassi e sono considerate magiche perché, alle loro pendici, molti hanno visto enormi pietre levitare al comando di potenti maghi."
The spaccatura nella roccia is a backdrop in the inferno di sassi. The description is "[mostra-entrata-grotta]La spaccatura sembra l[']ingresso ben nascosto di una grotta."

To say mostra-entrata-grotta:
[%] show-side figure of entrata-grotta; [%\$]
do nothing.

Section 1 - La grotta dei quaranta ladroni

The grotta dei quaranta ladroni is a female room. "L[']interno della grotta è molto ampio. Numerose [s]stalattiti e stalagmiti[x] l[']adornano, mentre la luce incerta che filtra dall[']esterno traccia sulle pareti ombre tremolanti dall[']aspetto minaccioso. Andando a [d]sud[x] esci dalla grotta[if the forziere is off-stage].[line break]Purtroppo, ti aspettavi di trovare montagne di tesori e invece non si vede nulla all[']interno, solo un po['] di [s]terra smossa[x] al centro della grotta[end if]." [%] The top-illustration is figure of grotta-dei-quaranta-ladroni. [%\$]

The stalattiti e stalagmiti is a backdrop in the grotta dei quaranta ladroni. "Le solite stalattiti e stalagmiti. Non sarebbe una grotta se non ci fossero."
The terra smossa is a backdrop in the grotta dei quaranta ladroni. "Solo un po['] di terra smossa, come se qualcuno avesse scavato nella grotta di recente."

Instead of going to the grotta dei quaranta ladroni when the player is carried by a rideable animal (called the mounted):
 say "[mostra-entrata-grotta][The mounted] cerca di attraversare l[']ingresso della grotta, ma è troppo stretto. L[']unico modo è entrare a piedi."

The forziere is a closed, locked, lockable container. "Vedi un [o]forziere[x] molto pesante, ti chiedi quanti tesori possa contenere." The description is "[mostra-forziere]Un pesante forziere [if open]aperto[otherwise]chiuso[end if]." The printed name is "[o]forziere[x]".

To say mostra-forziere:
[%] show-side figure of forziere; [%\$]
do nothing.

```
To say getta-forziere:
    [$%] show-side figure of forziere; [%$]
    [$%] play sound of caduta; [%$]
do nothing.
```

```
Instead of putting the forziere on the cammello:
    say "[mostra-dario-in-piedi][Nome-cammello] non sopporterebbe il peso del forziere. Devi trovare una bestia da soma."
```

The manciata di terra is a female thing. "La [o]terra[x] che hai scavato fa bella mostra di sé in un angolo della grotta." The description is "Un po['] di terra della grotta dei quaranta ladroni." The printed name is "manciata di [o]terra[x]".

```
Instead of digging in the grotta dei quaranta ladroni:
    if the player does not carry a digging-tool:
        say "Non puoi scavare questo terreno a mani nude.";
    otherwise if the manciata di terra is on-stage:
        say "Non c[']è più nulla da scavare qui.";
    otherwise:
        say "[mostra-forziere]Scavi dove la terra è più soffice e, grazie agli allenamenti che hai fatto, in poco tempo sei riuscito a scoprire un [o]forziere[x]. Lo trascini fuori dalla buca con grande fatica.";
        award 50 points;
        now the forziere is in the grotta dei quaranta ladroni;
        now the manciata di terra is in the grotta dei quaranta ladroni;
        now the capo dei ladroni is in the accesso alla grotta.
```

The capo dei quaranta ladroni is a man. "Vedi un gigante, il [o]capo[x] della famosa banda dei quaranta ladroni[if the capo dei quaranta ladroni carries the forziere]con il forziere sulle spalle[end if]." The description is "Il capo dei quaranta ladroni[if the capo dei quaranta ladroni carries the forziere]regge il forziere sulle spalle,[end if] è visibilmente infuriato e vuole vendicarsi dell[']oltraggio subito."

Understand "gigante" or "ladrone" as the capo dei quaranta ladroni.

```
Instead of doing anything when the capo dei quaranta ladroni is visible and the player carries the forziere:
    now the capo dei quaranta ladroni carries the forziere;
    say "Il capo dei quaranta ladroni ti sbarrà il passo, ti prende il forziere e ti sfida: 'Tu hai umiliato la mia abilità e la mia intelligenza scoprendo il nascondiglio dei miei tesori. Dimostrami di essere davvero più abile e intelligente di me e ti lascerò il forziere. Portami l[']arma o il gioco che preferisci, dalla scimitarra thailandese, agli scacchi indiani, fino al gioco del quindici; ma ricorda che se la tua abilità sarà inferiore alla mia... ti passerò a fil di spada.'[paragraph break]Rifletti tra te e te: 'E['] escluso che io sia più abile di questo gigante nel maneggiare le armi, quindi... quale gioco potrei portargli?'"
```

The cubo di Rubik can be fuori-gara, in-gara-aladino, in-gara-capo. The cubo di Rubik is fuori-gara.

```
Instead of giving something (called the object) to the capo dei quaranta ladroni:
    if the object is not the cubo di Rubik:
        say "Il ladrone risponde: 'Non so cosa farmene.[if the capo dei quaranta ladroni carries the forziere]Ti avevo chiesto un'arma da combattimento o un gioco d[']intelligenza. Vuoi sfidarmi con [an object]?[end if]'";
    otherwise if the cubo di Rubik is fuori-gara:
        say "Il ladrone prende il cubo e dice: 'Hai portato il famoso [o]cubo[x] di Rubik! Credi di essere più abile di me nel risolverlo? Fammi vedere quello che sai fare!' [line break]Dopo averlo rimescolato per qualche minuto te lo restituisce.";
        now the cubo di Rubik is mescolato;
        now the cubo di Rubik is in-gara-aladino;
    otherwise if the cubo di Rubik is in-gara-aladino:
        say "Il ladrone rifiuta il cubo dicendo: 'Prima voglio vedere cosa sai fare!";
    otherwise if the cubo di Rubik is in-gara-capo:
        if the cubo di Rubik is risolto:
            say "Il ladrone prende il [o]cubo[x] e dice: 'Ma è già risolto! Così è troppo facile: prima mescolalo... e mescolalo bene!' aggiunge sghignazzando, dopo avertelo restituito.";
        otherwise:
            award 75 points;
            now the forziere is in the accesso alla grotta;
            remove the capo dei quaranta ladroni from play;
            remove the cubo di Rubik from play;
            say "Il ladrone prende il cubo e comincia con perizia e incredibile velocità a ruotare le facce. Dopo cinque secondi ha quasi terminato e mancano solo una mezza dozzina di rotazioni, ma nella fretta invece di eseguire la combinazione ASA[']DAS[']A[']D['] esegue AS[']A[']D[']ASA[']D ed il cubo è più scombinato di prima. Le rotazioni si fanno frenetiche, ma la situazione peggiora a vista d[']occhio. Alla fine, dopo dieci minuti di marasma, il capo dei quaranta ladroni scaglia il cubo contro le rocce e lo frantuma in mille pezzi.[paragraph break]'Hai vinto,' dice demoralizzato, ma subito riprende con la sua solita superbia: 'Non capitarmi mai più tra le mani o farai una brutta fine!' e sparisce velocemente tra le rocce, dopo aver messo il [o]forziere[x] a terra.[getta-forziere]"
```

Instead of solving the cubo di Rubik when the capo dei quaranta ladroni is visible for the first time:

```

now the cubo is risolto;
now the cubo is in-gara-capo;
say "Con una sequenza di rapide rotazioni, risolvi il [o]cubo[x] in sette secondi netti. Il
ladrone dice ammirato: 'Incredibile! Sei davvero abile. Dammi subito il cubo! Ti faccio vedere io di
cosa sono capace!'"

```

The accesso alla grotta is a room. "Questo [s]dirupo[x] scosceso porta ad una grotta nel [s]fianco[x] della montagna. A [d]sud[x] si estende l[']inferno di [s]sassi[x], mentre a [d]nord[x] [if the porta di pietra is aperta]si apre la [s]spaccatura[x] nella roccia che conduce alla grotta[otherwise]un[']enorme [o]porta[x] di pietra ostruisce il passaggio tra le rocce che conduce alla grotta[end if]." [%\$] The top-illustration is figure of accesso-alla-grotta. [%\$]
The accesso alla grotta is south of the grotta dei quaranta ladroni.
The accesso alla grotta is north of the inferno di sassi.

The distesa di sassi is a backdrop in the accesso alla grotta.
The dirupo scosceso is a backdrop in the accesso alla grotta. "Le pendici della montagna si fanno scoscese. Bisogna proseguire a piedi."
The fianco della montagna is a backdrop in the accesso alla grotta. "Le pendici della montagna si fanno scoscese. Bisogna proseguire a piedi."
The cielo is a backdrop in the accesso alla grotta.
The montagne magiche is a backdrop in the accesso alla grotta.
The spaccatura nella roccia is a backdrop in the accesso alla grotta.

The porta di pietra is a female thing. The porta di pietra is in the accesso alla grotta. "[the init-desc of the porta di pietra]." The description is "[mostra-entrata-grotta]Solo un gigante o[if the porta di pietra is not stregata], forse,[end if] un mago potrebbero aver potuto [if the porta di pietra is aperta]smuovere questa pietra gigantesca[otherwise]collocare questa enorme pietra sull[']ingresso della grotta[end if]."

Instead of taking the porta di pietra:
say "Non sei abbastanza forte per sollevare la [o]porta[x] di pietra."

Instead of opening the porta di pietra:
say "Non sei abbastanza forte per spostare la [o]porta[x] di pietra."

To say the init-desc of the porta di pietra:
if the porta di pietra is aperta:
say "L[']ingresso della grotta è aperto. La pietra gigantesca che l[']ostruiva è stata rovesciata di lato";
otherwise:
say "Un[']enorme [o]pietra[x] ostruisce l[']entrata della grotta a nord. Ci vorrebbero cento uomini ed altrettanti cammelli per poterla smuovere. Ma ci dev[']essere anche un altro metodo[if porta di pietra is stregata] e ricordi di averlo visto in una visione[end if]".

The porta di pietra can be stregata. It is not stregata.
The porta di pietra can be aperta. It is not aperta.

Instead of going north when the location is the accesso alla grotta and the porta di pietra is not aperta:
say "[mostra-entrata-grotta]Purtroppo l[']accesso alla grotta è ostruito da una porta di pietra."

Instead of opening the porta di pietra:
say "[mostra-entrata-grotta]Non puoi spostare la pietra a mani nude, ma anche con qualche attrezzo l[']impresa sarebbe impossibile."

Opening magically is an action applying to nothing.

Understand "apriti sesamo" as opening magically.

Carry out opening magically:
if the location is not the accesso alla grotta:
say "Fiato sprecato! Cosa pensavi che succedesse?";
otherwise if the porta di pietra is aperta:
say "Fiato sprecato! La porta è già aperta.";
otherwise if the porta di pietra is stregata:
award 15 points;
[%\$]
play sound of caduta; [%\$]
say "[mostra-entrata-grotta]La terra comincia a tremare e, dopo qualche istante, l[']enorme porta di pietra cede di schianto scoprendo l[']entrata della grotta.";
now the porta di pietra is aperta;
otherwise:
say "Non succede nulla. Ci hai provato, ma non poteva essere così semplice."

The deserto di pietre is a luogo-lontano. It is south of the deserto di sabbia. The description is "Intorno a te riesci a vedere solo [s]pietre[x] rese aguzze dal vento, mentre a [d]nord[x] vedi in lontananza le [s]dune[x] del deserto di sabbia[if player is not carried by something], ma sono troppo lontane per arrivarci a piedi[end if]. A [d]sud[x] invece si estende a perdita d'occhio la [s]savana[x]." [%\$] The top-illustration is figure of deserto-di-pietre. [%\$]

The cielo is a backdrop in the deserto di pietre.

The distesa di pietre is a backdrop in the deserto di pietre. The description is "Pietre... pietre senza soluzione di continuità."

The dune di sabbia is a backdrop in the deserto di pietre.

The pianura della savana is a backdrop in the deserto di pietre. The description is "La piana della savana è costellata di rade acacie e bassi cespugli."

Chapter 5 - Il deserto di rocce

Section 1 - Il deserto di rocce

The deserto di rocce is luogo-lontano. It is east of the deserto di sabbia. The description is "Intorno a te riesci a vedere solo [s]rocce[x] squarciate dal caldo soffocante e rese affilate dal vento. A [d]ovest[x] vedi in lontananza le [s]dune[x] del deserto di sabbia[if player is not carried by something], ma sono troppo lontane per arrivarci a piedi[end if]; a [d]nord[x], tra le montagne, si scorge l'apertura di una stretta [s]gola[x]; mentre a [d]est[x] puoi ammirare la maestosa [s]valle[x] delle tombe." [%\$] The top-illustration is figure of deserto-di-rocce. [%\$]

The cielo is a backdrop in the deserto di rocce.

The distesa di rocce is a backdrop in the deserto di rocce. The description is "Rocce... rocce senza soluzione di continuità."

The pareti della gola nelle montagne is a backdrop in the deserto di rocce. The description is "Una stretta gola tra pareti a strapiombo."

The tombe nella valle is a backdrop in the deserto di rocce. The description is "L'immenso anfiteatro della valle delle tombe. Migliaia di anni di storia sono sepolti qui."

The dune di sabbia is a backdrop in the deserto di rocce.

Section 2 - Ingresso della gola

The ingresso della gola is a room. It is north of the deserto di rocce. "Sei giunto all'ingresso della gola. A [d]sud[x] si estende a perdita d'occhio il terrificante deserto di [s]rocce[x], mentre proseguendo a [d]est[x] ci si avventura in una stretta gola racchiusa tra due [s]pareti[x] scoscese. Qui i cammelli possono passare solo una volta ed è il luogo ideale per un'imboscata." [%\$] The top-illustration is figure of gola-tra-le-montagne. [%\$]

The distesa di rocce is a backdrop in the ingresso della gola.

The pareti della gola nelle montagne is a backdrop in the ingresso della gola.

Section 3 - La gola tra le montagne

The gola tra le montagne is a female room. It is east of the ingresso della gola. "[if unvisited]L'attraversamento della gola procede senza problemi. [end if]A [d]ovest[x] puoi vedere l'ingresso della gola, mentre a [d]est[x] se ne scorge l'uscita[x]. Per osservare il territorio circostante, puoi scalare le [s]pareti[x] e salire [d]su[x], in alto, sulla cima della montagna." [%\$] The top-illustration is figure of gola-tra-le-montagne. [%\$]

The ingresso alla gola nella montagna is a backdrop in the gola tra le montagne. The description is "L'ingresso è uno spiraglio di luce in lontananza."

The uscita alla gola nella montagna is a backdrop in the gola tra le montagne. The description is "L'uscita è uno spiraglio di luce in lontananza."

The pareti di rocce della gola della montagna is a female, plural-named backdrop in the gola tra le montagne. The description is "Le pareti sono molto ripide. Puoi scalarle, ma con grande difficoltà."

Section 4 - Nido dell'avvoltoio

Understand "alto" or "in alto" as up.

Understand "basso" or "in basso" as down.

The pareti di rocce della gola della montagna is in the cima della montagna.

The cima della montagna is a female room. It is up of the gola tra le montagne. "La scalata è stata dura, ma ne valeva la pena. La vista da quassù è incredibile[if unvisited] e puoi osservare il paesaggio dal deserto di [s]sabbia[x] che vedi in lontananza verso ovest, fino al grande [s]fiume[x] ad est. Laggiù potrai sicuramente imbarcarti su un traghetto che ti riporterà finalmente a casa[end if]. Da qui puoi vedere l'accampamento[x] dei predoni, che sembra [if the orda di predoni is in the accampamento dei predoni]molto ben presidiato. Si notano molte [s]sentinelle[x] lungo tutto il perimetro e un'intensa attività intorno ai fuochi. E' impossibile sperare di passare inosservati. Meglio aspettare un altro momento per tentare di raggiungere il fiume a est[otherwise]completamente sguarnito. I predoni devono essere usciti in massa ed ora rimangono solo poche [s]sentinelle[x] intorno ai fuochi. E' il momento buono per tentare di passare[end if]. [line break]Scendi [d]giù[x] per tornare nella gola." [%\$] The top-illustration is figure of cima-della-montagna. [%\$]

The cielo is a backdrop in the cima della montagna.
 The dune di sabbia is a backdrop in the cima della montagna.
 The meandri del fiume is a backdrop in the cima della montagna. The description is "I meandri del fiume si contorcono da sud verso nord, trasportando acqua in grande quantità. E['] la vita per tutta la regione."
 The predoni in accampamento is a backdrop in the cima della montagna. The description is "L[']accampamento dei predoni. Numerose tende difese da un recinto."
 The sentinelle dei varchi is a backdrop in the cima della montagna.

After going to the cima della montagna for the first time:
 award 10 points;
 now the accampamento dei predoni is osservato;
 continue the action.

After going to the cima della montagna for more than one time:
 now the accampamento dei predoni is osservato;
 continue the action.

Instead of entering when the pareti di rocce della gola della montagna are visible:
 try going up.

Instead of climbing the pareti di rocce della gola della montagna:
 try going up.

Instead of getting off the pareti di rocce della gola della montagna:
 try going down.

Instead of exiting when the pareti di rocce della gola della montagna are visible:
 try going down.

Instead of going up when the player is carried by a rideable animal (called the mounted):
 say "[The mounted] cerca di risalire le pareti della gola, ma sono troppo scoscese.
 [d]Scendi[x] prima di tentare la salita a piedi."

Before going down when the player is in the cima della montagna:
 say "Non l[']avresti mai detto, ma la discesa è stata più difficile della salita. Non avevi alcun dubbio di riuscire a raggiungere il fondo, ma temevi di arrivarci... troppo in fretta!";
 continue the action.

The nido is an open container in the cima della montagna.

Instead of taking the nido:
 say "Non puoi prendere il nido, perché è cementato alla roccia dal guano."

The uovo is an edible thing. The uovo is in the nido. The description is "Un grosso uovo d[']avvoltoio. Chissà che gusto ha?" The printed name is "[o]uovo[x]".

Instead of eating the uovo for the first time:
 say "Questo è l[']unico uovo d[']avvoltoio nel raggio di decine di miglia. Pensi davvero che sia stato covato amorevolmente per settimane solo per fartelo mangiare? Quest[']uovo ha un altro scopo nella vita. Ti ho avvertito!"

Instead of eating the uovo for more than one time:
 now the uovo is in the nido;
 now the avvoltoio is in the cima della montagna;
 now the avvoltoio is dissetato;
 say "[mostra-attacco-avvoltoio]Tenti di rompere il guscio, ma non ci riesci, così lo sollevi sopra la testa e stai per scagliarlo contro una roccia quando, improvvisamente, ti assale l[']avvoltoio e con gli artigli ti sottrae l[']uovo. Poi, con la stessa rapidità con cui ti ha aggredito, si allontana dopo essersi ripreso il maltolto."

To say mostra-attacco-avvoltoio:
 [%] show-side figure of avvoltoio; [%]
 [%] play sound of avvoltoio-attacco; [%]
 do nothing.

To say mostra-volo-avvoltoio:
 [%] show-side figure of avvoltoio; [%]
 [%] play sound of avvoltoio-in-volo; [%]
 do nothing.

Instead of doing something except examining with the uovo when the uovo is in the nido and the avvoltoio is in the cima della montagna:
 say "E['] un ottimo sistema per mettere alla prova i sentimenti materni di un uccello. [run paragraph on]";
 try touching the avvoltoio.

Instead of putting a thing (called the carico) on the gobba:

```

if the carico is a cosa-ingombrante:
    continue the action;
otherwise:
    say "Non puoi. L[']equilibrio sarebbe precario e cadrebbe giù."

```

```

After inserting the uovo into the tasche for the first time:
    award 50 points;
    continue the action.

```

The avvoltoio is an animal. It is in the deserto di sabbia. "Vedi anche un [The init-desc of the avvoltoio]. The description is "[mostra-volo-avvoltoio]L[']avvoltoio ha un aspetto inquietante. Ti fissa come se ti conoscesse e fosse lì proprio per te."

The avvoltoio can be assetato or dissetato. The avvoltoio is assetato.

```

To say the init-desc of the avvoltoio:
    if the location is the deserto di sabbia:
        say "[o]avvoltoio[x]. Sta volando molto in alto descrivendo lentamente degli
invisibili cerchi nel [s]cielo[x]. Sembra che si stia spostando [if the avvoltoio is assetato]a
ovest verso l[']oasi, probabilmente per abbeverarsi[otherwise]a est verso le montagne, probabilmente
sta tornando al nido[end if].[run paragraph on]";
        otherwise if the location is the sorgente dell'oasi:
            say "avvoltoio. Si sta abbeverando, ma appena ti scorge, dopo aver faticosamente
preso la rincorsa e agitato le lunghe e pesanti ali, spicca il volo ed in pochi istanti è già
lontano.[run paragraph on]";
            move the avvoltoio to the deserto di sabbia;
        otherwise if the location is the cima della montagna:
            say "[o]avvoltoio[x]. E['] intento a consolidare il proprio nido. Non avresti mai
detto che un animale dall[']aspetto così truce potesse essere capace di sentimenti materni.[run
paragraph on]";
            now the avvoltoio is assetato.

```

Instead of doing something except examining to the avvoltoio when the location is the cima della montagna:

```

say "[mostra-attacco-avvoltoio]L[']avvoltoio, [one of]con un guizzo fulmineo della sua testa
pelata[or]dopo aver spiegato le sue lunghe ali[or]mostrandoti gli artigli affilati[at random], ti
colpisce con il suo becco acuminato. Non ti conviene insistere, avresti sicuramente la peggio."

```

Section 5 - Uscita della gola

```

To say rapporto sull'accampamento:
    if accampamento dei predoni is osservato:
        say ", ma da quello che hai visto poco fa, è [if the orda di predoni is in the
accampamento dei predoni]ben presidiato[otherwise]completamente sguarnito[end if]";
    otherwise:
        say ", ma al centro della gola ci dovrebbe essere un passaggio per salire in cima
alla montagna ed osservare il territorio circostante".

```

The uscita della gola is a female room. It is east of the gola tra le montagne. "L[']uscita della gola si apre a [d]sud[x] sull[']accampamento[x] dei predoni, mentre a [d]ovest[x] puoi addentrarti all[']interno della [s]gola[x]. Da qui non puoi osservare i movimenti nell[']accampamento[rapporto sull'accampamento]." [%] The top-illustration is figure of gola-tra-le-montagne. [%]

The pareti della gola nelle montagne is a backdrop in the ingresso della gola.

Chapter 6 - La valle delle tombe

Section 1 - La valle delle tombe

The valle delle tombe is a female room. It is east of the deserto di rocce. "Qui comincia la valle delle tombe.[if unvisited] Sin da tempi immemorabili i re della regione si facevano costruire ampie dimore in cui giacere per l[']eternità, ovviamente dopo la loro morte. Questa almeno era la loro intenzione, ma nessuna tomba è mai rimasta inviolata se non per pochi anni. Ora la [s]valle[x] è spesso occupata da orde di predoni che l[']attraversano per assalire le vie carovaniere.[end if] [line break]A [d]ovest[x] vedi il deserto di [s]rocce[x], mentre ad [d]est[x] c[']è l[']accampamento[x] dei predoni: devi passargli accanto se vuoi raggiungere il fiume." [%] The top-illustration is figure of valle-delle-tombe. [%]

The cielo is a backdrop in the valle delle tombe.
The tombe nella valle is a backdrop in the valle delle tombe.
The distesa di rocce is a backdrop in the valle delle tombe.
The predoni in accampamento is a backdrop in the valle delle tombe.

Section 2 - L'accampamento dei predoni

The accampamento dei predoni is a room. It is east of the valle delle tombe. It is south of the uscita della gola. "Ti trovi lungo il perimetro dell[']accampamento dei predoni. A [d]ovest[x] vedi la [s]valle[x] delle tombe, ad [d]est[x] si vedono le rive del grande [s]fiume[x], mentre a

```
[d]nord[x], tra le montagne, si scorge l[']apertura di una stretta [s]gola[x]. L[']
[s]accampamento[x] è [if the orda di predoni is in the accampamento dei predoni]molto ben presidiato
e le numerose [s]sentinelle[x] poste all[']entrata dei varchi potrebbero notare la tua presenza
molto presto. Se indugi troppo a lungo, i primi drappelli ti assaliranno e non è sicuro che tu possa
scamparla questa volta[otherwise]sguarnito, ma le poche [s]sentinelle[x] rimaste avrebbero comunque
la meglio su di te se indugi troppo[end if]." [%$] The top-illustration is figure of accampamento-
dei-predoni. [%$]
```

```
The cielo is a backdrop in the accampamento dei predoni.
The tombe nella valle is a backdrop in the accampamento dei predoni.
The predoni in accampamento is a backdrop in the accampamento dei predoni.
The meandri del fiume is a backdrop in the accampamento dei predoni.
The pareti della gola nelle montagne is a backdrop in the accampamento dei predoni.
The sentinelle dei varchi is a backdrop in the accampamento dei predoni. The description is "Feroci
predoni armati di tutto punto presidiano i varchi."
```

```
After going to the accampamento dei predoni:
    now the accampamento dei predoni is not osservato;
    continue the action.
```

```
The accampamento dei predoni can be osservato. The accampamento dei predoni is not osservato.
```

```
The orda di predoni is a female person in the accampamento dei predoni. "[if the location is the
accampamento dei predoni]Questi [o]predoni[x] non sembrano così minacciosi quando sono impegnati
nelle loro faccende quotidiane[otherwise]L['] [o]orda[x] di predoni ti insegue per farti la pelle una
volta per tutte. E['] bene darsela a gambe![end if]". The description is "[if the location is the
accampamento dei predoni]Cento predoni impegnati nelle faccende domestiche: alcuni cucinano, altri
strigliano i cammelli, altri ancora stanno facendo i conti per dividere l[']ultimo bottino. Chi
l[']avrebbe mai detto?[otherwise][mostra-orda]Cento predoni urlanti e scatenati, ma soprattutto
cento cammelli inferociti: non si possono trovare due ragioni migliori per darsela a gambe.[end if]"
```

```
To say mostra-orda:
    [%$] show-side figure of orda-di-predoni; [%$]
    do nothing.
```

```
Instead of doing anything except taking inventory or examining or looking when the location is the
accampamento dei predoni and the orda di predoni is in the accampamento dei predoni:
    if the room gone to is the valle delle tombe:
        say "[mostra-uomo-e-dario-2]Sei riuscito a passare inosservato. Nessuno ti insegue.";
    otherwise:
        say "[mostra-orda]La tua azione non è stata molto discreta e sei stato individuato.
Le sentinelle ti sono subito addosso e sei costretto a tornare nella valle delle tombe inseguito da
decine di predoni urlanti a dorso di cammello.";
        move the orda di predoni to the valle delle tombe;
        if the player is carried by a rideable animal (called the mounted):
            move the mounted to the valle delle tombe;
            if the asino is tied:
                move the asino to the valle delle tombe;
            try looking;
        otherwise if the cammello is in the valle delle tombe and the orda di predoni is in the
valle delle tombe:
            now the player is carried by the cammello;
            say "[mostra-dario-in-piedi]Salti di corsa sul dromedario, che ti attendeva nella
valle, e ti lanci al galoppo[if the asino is tied] seguito dall[']asino[end if].";
        otherwise:
            move the player to the valle delle tombe;
            if the asino is in the valle delle tombe and the asino is untied and the forziere is on the
dorso and the orda di predoni is in the valle delle tombe:
                award 25 points;
                say "[mostra-asino]I predoni notano l[']asino carico del forziere e quindi si gettano
al suo inseguimento per impossessarsi del tesoro.[line break]L[']asino, alla vista dei predoni
urlanti con i loro cammelli grugnenti, cambia immediatamente abitudini e comincia a galoppare
spontaneamente per tutta la valle delle tombe, inseguito dall[']orda.";
            otherwise if the forziere is in the valle delle tombe and the orda di predoni is in the
valle delle tombe:
                say "Fortunatamente, i predoni non hanno notato il forziere seminascolato nella
sterpaglia, altrimenti se ne sarebbero subito impossessati, sarebbero tornati trionfanti alle loro
tende e non avresti più avuto un[']esca per fargli abbandonare l[']accampamento."
```

```
Instead of doing anything except taking inventory or examining or looking when the location is the
valle delle tombe and the orda di predoni is in the valle delle tombe:
    say mostra-uomo-e-dario-2;
    if the room gone to is the deserto di rocce:
        say "Sei riuscito a seminare l[']orda di predoni.";
    otherwise:
        say "Dopo un lungo inseguimento, sei riuscito a seminare l[']orda di predoni[if the
asino is untied and the asino is in the valle delle tombe], che è anche impegnata a rincorrere
l[']asino con il suo forziere per tutta la valle delle tombe[end if]. Purtroppo, nonostante le
alternative che hai tentato, l[']unico sistema efficace è stato quello di addentrarsi nel deserto di
```

```

rocce.";
  if the player is carried by a rideable animal (called the mounted):
    move the mounted to the deserto di rocce;
    if the asino is tied:
      move the asino to the deserto di rocce;
    try looking;
  otherwise:
    move the player to the deserto di rocce.

```

```

To say mostra-uomo-e-dario-1:
  [%] show-side figure of uomo-e-dario-1; [%]
do nothing.

```

```

To say mostra-uomo-e-dario-2:
  [%] show-side figure of uomo-e-dario-2; [%]
do nothing.

```

```

After going from the deserto di rocce when the orda di predoni is in the valle delle tombe and (the asino is not in the valle delle tombe or the forziere is not on the dorso):
  if the room gone to is not the valle delle tombe:
    move the orda di predoni to the accampamento dei predoni;
  continue the action.

```

```

Instead of waiting when the location is the deserto di rocce and the orda di predoni is in the valle delle tombe and (the asino is not in the valle delle tombe or the forziere is not on the dorso):
  move the orda di predoni to the accampamento dei predoni;
  continue the action.

```

Section 3 - Le riva sinistra del fiume

The riva sinistra del fiume is a female room. It is east of the accampamento dei predoni. "Ecco le rive del grande fiume che scorre da sud a nord. A [d]ovest[x] c[']è l[']accampamento[x] dei predoni, mentre ad est[if unvisited] inizia la via per tornare a casa... è un peccato che di mezzo ci sia il grande [s]fiume[x]. La corrente è troppo forte per attraversare a nuoto e non ci sono guadi in questa stagione[otherwise], oltre il fiume, inizia la via per tornare a casa[end if]. L[']unico modo per passare è imbarcarsi sul traghetto." [%] The top-illustration is figure of riva-sinistra-del-fiume. [%\$]

```

The cielo is a backdrop in the riva sinistra del fiume.
The predoni in accampamento is a backdrop in the riva sinistra del fiume.
The meandri del fiume is a backdrop in the riva sinistra del fiume.

```

The zattera di tronchi is a female enterable supporter. The keyword is "zattera". The zattera di tronchi is in the riva sinistra del fiume. "Il traghetto non è altro che una [o]zattera[x] molto instabile legata ad una [o]corda[x] tesa da una riva all[']altra e governata da un lungo [o]remo[x]. Non sembra molto sicura, ma non c[']è altro mezzo". The description is "[mostra-zattera]Una zattera di tronchi legata ad una [o]corda[x] con un lungo [o]remo[x]."

```

To say mostra-zattera:
  [%] show-side figure of zattera; [%]
do nothing.

```

The barile pieno is a closed, openable container. The keyword is "barile". The barile pieno is on the zattera di tronchi. The description is "Un barile pieno d'acqua [if open]aperto[otherwise]chiuso[end if]". The printed name is "[o]barile[x]". The pesce is a thing in the barile pieno. The description is "Un pesce che guizza nell[']acqua. Ti fa compassione così imprigionato." The printed name is "[o]pesce[x]".

```

Instead of taking the barile pieno:
  say "Il barile è inchiodato alla zattera."

```

```

Instead of taking the pesce when the traghettatore is on-stage:
  now the barile pieno is closed;
  say "'Ehi!' esclama il traghettatore, 'non toccare la mia cena!' e spazientito chiude il barile."

```

```

Instead of taking the pesce when the traghettatore is off-stage:
  award 5 points;
  remove the pesce from play;
  [%] show-side figure of pesce; [%]
  [%] play sound of tuffo; [%]
  say "Ci vogliono alcuni tentativi, ma alla fine riesci ad afferrare il pesce e lo getti fuori bordo. Dopo qualche istante, in lontananza, lo vedi guizzare felice tra i flutti del fiume."

```

Understand "traghetto" as zattera di tronchi.

A cosa-remante is a kind of thing. A cosa-remante can be spinto or tirato. A cosa-remante is usually spinto. A cosa-remante has a number called the pagaiate.

The lungo remo is a cosa-remante. The lungo remo has pagaiate 0. The lungo remo is part of the zattera di tronchi. The description is "Un lungo remo. Spingendolo e tirandolo si governa il traghettatore."

The corda tesa is a thing. The corda tesa is part of the zattera di tronchi. The description is "Una robusta corda di peli di cammello, tesa da una riva all[']altra."

Instead of cutting the corda tesa:

```
say "Non conviene tagliare la corda... in senso letterale. Il traghettatore verrebbe trascinato via dalla corrente."
```

The traghettatore is a man in the riva sinistra del fiume. "Vedi un [o]barcaiolo[x] [if the traghettatore is the current interlocutor]che sta parlando con te[otherwise]intento ad avvolgere le [s]gomene[x] della sua imbarcazione.[end if]. Ha l[']aspetto di un pirata con quella benda nera sull[']occhio sinistro ed un loquace [o]pappagallo[x], che si chiama Cocorito, sulla spalla destra." The description is "[mostra-pappagallo]Un vero pirata con il suo fido pappagallo[if the traghettatore is the current interlocutor] che sta parlando con te[otherwise]. Se vuoi parlargli, prova prima a salutarlo con un [t]ciao[x][end if]."

To say mostra-pappagallo:

```
[%] show-side figure of pappagallo; [%]
do nothing.
```

The gomene is a female, plural-named scenery in the riva sinistra del fiume. The description is "Robuste corde di... peli di cammello."

Understand "pirata" or "barcaiolo" as traghettatore.

The pappagallo is an animal. The pappagallo is part of the traghettatore. The description is "[mostra-pappagallo]Un bel pappagallo variopinto. Sembra che abbia molta più voglia di parlare del suo padrone.Se vuoi parlargli, prova a salutarlo dicendo: '[t]Cocorito, ciao[x]'."

Understand "Cocorito" as pappagallo.

The lavoro TRG is a discussable. The printed name is "[t]lavoro[x]".

The barca TRG is a discussable. The printed name is "[t]barca[x]".

The denaro TRG is a discussable. The printed name is "[t]denaro[x]".

The lavoro TRG, barca TRG, denaro TRG are part of the traghettatore.

traghettatore-node is a closed convnode.

The node of the traghettatore is traghettatore-node.

The ask-suggestions are {lavoro TRG}.

The tell-suggestions are {barca TRG}.

After saying hello to the traghettatore when the greeting type is explicit:

```
say "'Salute traghettatore!' esordisci.[line break]Il traghettatore interrompe il [t]lavoro[x] sulla sua [t]barca[x], ti osserva e dice: 'Salute a te!'"
```

After saying hello to the traghettatore when the greeting type is implicit:

```
say "Attiri l[']attenzione con un colpo di tosse. Il traghettatore si volta e attende che tu inizi a parlare."
```

After saying goodbye to the traghettatore when the farewell type is explicit:

```
say "'Bene, allora ti saluto!' concludi.[line break]'Alla prossima,' risponde e torna al suo lavoro."
```

After saying goodbye to the traghettatore when the farewell type is implicit:

```
say "Il traghettatore ti saluta e torna a lavorare."
```

Default Response of traghettatore-node:

```
say "'Se non hai denaro per pagare la traversata, è inutile che tu mi faccia perdere tempo in chiacchiere,' dice il traghettatore con impazienza.[line break]";
try silently listing suggested topics.
```

Response of traghettatore-node when asked about the lavoro TRG:

```
say "'Ti occupi da molto tempo di navigazione?' chiedi al traghettatore.[paragraph break]Ti risponde: 'Da sempre! Ho navigato i sette mari ed i cinque oceani, ho attraversato uragani, tempeste e cicloni. Sopra di me hanno soffiato i venti più impetuosi, sotto di me sono passate le onde più alte. Ti potrei raccontare di mostri marini, tesori nascosti, navi fantasma, ma ora non ho tempo...' e proprio sul più bello smette di raccontare.[paragraph break]Gli replichi: 'Anch[']io non ho tempo da perdere: devo attraversare il fiume!'[paragraph break]'Allora dovrai trovarlo questo tempo da perdere, perché a giudicare da come sei combinato, non hai [t]denaro[x] per pagare il passaggio. Su questo non ho altro da dire.'[remove lavoro TRG ask suggestion][add-lavoro-suggestion-pappagallo][add denaro TRG ask suggestion]"
```

To say add-lavoro-suggestion-pappagallo:

```
add lavoro TRG to ask-suggestions of the node of the pappagallo.
```

Response of traghettatore-node when asked about the denaro TRG:

```
say "'E['] vero non ho denaro! Non sai dove potrei trovarne?' chiedi ingenuamente.[paragraph
break]Ti risponde ironico fissandoti con l[']unico occhio rimasto: 'Certo che so dove potresti
trovarne! Ma di sicuro non ti dirò dove tengo i miei risparmi.' Dopo aver riso sguaiatamente
conclude: 'E ora non ho davvero più nulla da dire.'[remove denaro TRG ask suggestion]"
```

Response of traghettatore-node when told about barca TRG:

```
say "'Un traghetto davvero molto instabile,' affermi senza mezzi termini.[paragraph break]'Se trovi
di meglio a meno di dieci miglia a monte o a valle,' risponde ironico il traghettatore, 'accomodati
pure.'[add-barca-suggestion-pappagallo][remove barca tell suggestion]"
```

To say add-barca-suggestion-pappagallo:

```
add barca TRG to tell-suggestions of the node of the pappagallo.
```

pappagallo-node is a closed convnode.

The node of the pappagallo is pappagallo-node.

After saying hello to the pappagallo when the greeting type is explicit:

```
say "[mostra-pappagallo]'Salute Cocorito!' esordisci.[line break]Il pappagallo ti risponde: 'Salute
Cocorito, crack!'"
```

After saying hello to the pappagallo when the greeting type is implicit:

```
say "[mostra-pappagallo]Attiri l[']attenzione con un fischio. Il pappagallo ti fissa con un occhio e
fischia a sua volta."
```

After saying goodbye to the pappagallo when the farewell type is explicit:

```
say "'Addio Cocorito!' concludi.[line break]'Addio Cocorito, crack!' risponde e torna a pulirsi le
penne."
```

After saying goodbye to the pappagallo when the farewell type is implicit:

```
say "Il pappagallo ricomincia a pulirsi le penne."
```

Default Response of pappagallo-node:

```
say "'Salute Cocorito! Crack!' dice il pappagallo.[line break]";
try silently listing suggested topics.
```

Response of pappagallo-node when asked about the lavoro TRG:

```
say "'Il tuo padrone si occupa da molto tempo di navigazione?' chiedi al pappagallo.[paragraph
break]Ti risponde: 'Da sempre! Crack! Ha navigato i sette mari ed i cinque oceani, ha attraversato
uragani, tempeste e cicloni. Crack! Sopra di lui hanno soffiato i venti più impetuosi, sotto di lui
sono passate le onde più alte. Crack! Ti potrebbe raccontare di mostri marini, tesori nascosti, navi
fantasma, ma ora non ha tempo...'[paragraph break]Gli replichi: 'Ma io devo attraversare il
fiume!'[paragraph break]'Allora aspetterai un pezzo, crack!, perché a giudicare da come sei
combinato, crack!, non hai [t]denaro[x] per pagare il passaggio. Su questo non ha altro da dire.
Crack!'[add denaro TRG ask suggestion][remove lavoro TRG ask suggestion]"
```

Response of pappagallo-node when asked about the denaro:

```
award 50 points;
say "[remove denaro TRG ask suggestion]Chiedi maliziosamente al pappagallo: 'Non ho denaro!
Non sai dove potrei trovarne?'[paragraph break]Cocorito ti risponde simulando un tono ironico e
fissandoti con un occhio: 'Certo che sa dove potresti trovarne! Crack! Ma di sicuro non ti dirà che
tiene i suoi risparmi - crack! - sotto al letto nella baracca in cima alla collina!
Crack!'[paragraph break]'Brutta bestiaccia,' urla il traghettatore, 'devi sempre spifferare tutti i
fatti miei? Ora devo subito cambiare nascondiglio!'[paragraph break]Si avvia verso la sua
[s]baracca[x] salendo di corsa lungo il fianco della collina e in pochi istanti sparisce con il suo
pappagallo dalla lingua lunga.[paragraph break]Ne approfitti immediatamente per sciogliere la cima
d[']ormeggio. [no line break]";
if the player is not on the zattera di tronchi:
    try silently entering the zattera di tronchi;
    [%]
    show-top figure of grande-fiume; [%]
    say "[mostra-zattera]La corrente impetuosa trascina subito la zattera in mezzo al fiume e lì
si ferma.";
    remove traghettatore from play;
    now the zattera di tronchi is in the grande fiume.
```

The baracca in cima alla collina is a female backdrop in the riva sinistra del fiume. The description is "Una baracca che si intravede in lontananza, in cima alla collina."

Response of pappagallo-node when told about barca:

```
say "'Un traghetto davvero molto instabile,' dici al pappagallo senza peli sulla lingua.[paragraph
break]'Se trovi di meglio, crack!, a meno di dieci miglia a monte o a valle,' risponde il pappagallo
simulando un tono ironico, 'accomodati pure, crack!'[remove barca TRG tell suggestion]"
```

Instead of entering the zattera di tronchi:

```
say rumore-onde;
if the player is carried by a rideable animal (called the mounted):
    now the mounted is on the zattera di tronchi;
    say "[mostra-dario-in-piedi][Nome-cammello] salta sulla zattera ed i tronchi, legati
strettamente, scricchiolano in modo pauroso sotto le sue zampe.";
```

```

otherwise:
    say "[mostra-zattera]Salti sulla zattera ed i tronchi, legati strettamente,
scricchiolano in modo preoccupante al tuo passaggio.";
    now the player is on the zattera di tronchi.

```

```

Instead of getting off the zattera di tronchi:
    say rumore-onde;
    if the player is carried by a rideable animal (called the mounted):
        now the mounted is in the location;
        say "[mostra-dario-in-piedi][Nome-cammello] salta sulla riva e la zattera ondeggia
paurosamente.";
    otherwise:
        say "[mostra-zattera]Salti sulla riva e la zattera ondeggia vistosamente.";
        now the player is in the location.

```

```

Instead of doing something except examining with the lungo remo when the traghettatore is on-stage:
    say "Il traghettatore ti deride: 'Lascia stare! E['] inutile che provi a manovrare il remo:
ci vogliono anni di apprendistato per riuscire a tirarlo e spingerlo come si deve.'"

```

Section 4 - Il grande fiume

The grande fiume is a room. It is east of the riva sinistra del fiume. "Sei in mezzo al fiume, dove la corrente è più forte. A est vedi la [s]riva[x] destra, dove inizia la via delle carovane. Da nord scende inesorabile l[']impetuosa [s]corrente[x] del fiume, mentre poche miglia a sud si vede il vapore alzato dalla prima [s]cataratta[x]." [%\$] The top-illustration is figure of grande-fiume. [%\$]

```

To say rumore-onde:
    [%$] play sound of onde; [%$]
    do nothing.

```

The cielo is a backdrop in the grande fiume.
The impetuosa corrente is a backdrop in the grande fiume. "[rumore-onde]L[']acqua scorre turbolenta. E['] impossibile nuotarci dentro: verresti risucchiato dai gorgi."
The prima cataratta is a backdrop in the grande fiume. "[rumore-onde]Una cascata rovinosa. La zattera verrà sicuramente sconquassata se ci finirà dentro."
The riva destra is a backdrop in the grande fiume. "La meta è ormai vicina. La riva destra è lì che ti aspetta."
The baracca in cima alla collina is a backdrop in the grande fiume.

```

Instead of exiting when the location is the grande fiume:
    Say "Se scendi dalla zattera, sarai travolto dai gorgi impetuosi del grande fiume e
affogherai."

```

```

Before pushing the lungo remo when the traghettatore is off-stage and the lungo remo is spinto:
    [%$] play sound of caduta; [%$]
    say "Il lungo remo è già spinto a fine corsa." instead.

```

```

Before pulling the lungo remo when the traghettatore is off-stage and the lungo remo is tirato:
    [%$] play sound of caduta; [%$]
    say "Il lungo remo è già tirato a fine corsa." instead.

```

```

Instead of pulling the lungo remo when the traghettatore is off-stage:
    now the lungo remo is tirato;
    increase the pagaiate of the lungo remo by 1;
    if the pagaiate of the lungo remo is 1:
        say "[rumore-onde]Provi a tirare il remo e, con grande fatica, riesci a spostare la
zattera, ma di molto poco. Pensi: 'Il traghettatore aveva ragione! Per manovrare questa zattera
bisogna aver navigato tutti i sette mari ed i cinque oceani!";
        otherwise if the pagaiate of the lungo remo is 3:
            say "[rumore-onde]Provi ancora a tirare il remo. Ormai hai preso padronanza. E[']
quasi fatta!";
        otherwise if the pagaiate of the lungo remo is greater than 4:
            say "La zattera è ormeggiata."

```

```

Instead of pushing the remo when the traghettatore is off-stage:
    now the lungo remo is spinto;
    increase the pagaiate of the lungo remo by 1;
    if the pagaiate of the lungo remo is 2:
        say "[rumore-onde]Non ti perdi d[']animo e provi a spingere il remo. Con meno fatica
di prima, la zattera si sposta leggermente verso la riva destra. Pensi: 'Certo! Venti impetuosi
hanno soffiato sopra il traghettatore e sotto di lui sono passate le onde più alte, ma - per tutte
le navi del deserto! - questo fiume sarà anche grande, però è solo un fiume!";
        otherwise if the pagaiate of the lungo remo is 4:
            now the zattera di tronchi is in the via delle carovane;
            say "[rumore-onde]Con olio di gomito governi la zattera che attracca, anche se con
qualche difficoltà, alla riva destra. Pensi: 'Mi sono perso i mostri marini, i tesori nascosti e le
navi fantasma... ma ho attraversato il fiume!";
            try looking;

```

otherwise if the pagaiate of the lungo remo is greater than 4:
say "La zattera è ormeggiata."

Section 5 - La via delle carovane

The via delle carovane is a female room. "La via delle carovane, battuta per secoli da innumerevoli cammelli, si apre di fronte a te ampia e libera. Attraverso di essa tutte le mercanzie più preziose e ricercate raggiungono i più remoti angoli del mondo, facendo la fortuna di coraggiosi mercanti.
[line break]Ad [d]ovest[x] vedi i meandri del [s]fiume[x], mentre ad [d]est[x] si apre la via di [s]casa[x]." [%\$] The top-illustration is figure of via-delle-carovane. [%\$]

The cielo is a backdrop in the via delle carovane.
The meandri del fiume is a backdrop in the via delle carovane.
The via di casa is a backdrop in the via delle carovane.

Instead of exiting when the player is on the zattera di tronchi and the location is the via delle carovane:

```
say "[rumore-onde]Con un salto [if the player is carried by the cammello][mostra-dario-in-piedi][nome-cammello] scende[otherwise][mostra-uomo-e-orme]scendi[end if] dal traghetto e subito ti senti rassicurato nel percepire la terraferma sotto di te, ma soprattutto sei rinfrancato dal fatto che questa brutta avventura sia finita.[paragraph break][happy-end]";
end the game in victory
```

Instead of getting off the zattera di tronchi when the location is the via delle carovane:

```
say rumore-onde;
if the player is carried by a rideable animal (called the mounted):
  now the mounted is in the location;
  say "[mostra-dario-in-piedi]Con un salto, [nome-cammello] scende dal traghetto. [no line break]";
otherwise:
  say "[mostra-uomo-e-orme]Con un salto scendi dal traghetto. [no line break]";
  now the player is in the location;
say "Subito ti senti rassicurato nel percepire la terraferma sotto di te, ma soprattutto sei rinfrancato dal fatto che questa brutta avventura sia finita.[paragraph break][happy-end]";
end the game in victory.
```

To say happy-end:

```
[%$] play sound of colonna sonora; [%$]
say "[happy-commento][happy-coda]".
```

To say happy-commento:

```
say "[line break]Certo! E['] stato difficile arrivare fino a qui e non ci hai guadagnato nulla, ma anche questa è stata un[']esperienza, una delle tante - gratificanti o meno - che hai vissuto.[paragraph break]Non posso che farti i complimenti per aver completato l[']impresa e spero, comunque, che anche per te che l[']hai percorsa dall[']inizio alla fine sia stata in fondo una bella avventura, come lo è stata per me che l[']ho scritta.[line break]Quando ho iniziato non avevo idea di dove sarebbe andata a parare: c[']era solo Aladino, il mercante sperduto nel deserto di sabbia. Poi, lentamente, sono comparsi intorno a lui tanti personaggi, con le loro caratteristiche e manie, sono sorti nuovi paesaggi e strade da percorrere.[paragraph break]Come dicevo, almeno per me, è stata una bella avventura da portare a termine e, se lo è stata anche per te, ti consiglio di non perdere '[bold type]Aladino ed i Pirati Saraceni[roman type]', il seguito di 'Aladino e le Perle di Saggezza'... prossimamente su questi schermi!"
```

To say happy-coda:

```
say "[line break] '[bold type]Aladino e le Perle di Saggezza[roman type]' [line break]una storia da [italic type]Mille e Una Avventura[roman type] [line break]una produzione di [italic type]Leonardo Boselli[roman type] [line break]scritta con Inform7 di Graham Nelson et al. [paragraph break]Mille e Una Avventura: [bold type]http://milleuna.sf.net[roman type] [line break]".
```

Rule for amusing a victorious player:

```
say "[elenca-proverbi]".
```

Chapter 7 - Aperta savana

The aperta savana is a female room. It is south of the deserto di pietre. "Sei giunto nella savana. Il paesaggio è più regolare ed è ricoperto da radi alberi di [s]acacia[x] e cespugli spinosi.[line break]A [d]nord[x] si estende il grande deserto di [s]pietre[x], mentre a [d]sud[x] si vedono le [s]capanne[x] del villaggio degli agricoltori." [%\$] The top-illustration is figure of aperta-savana. [%\$]

The cielo is a backdrop in the aperta savana.
The distesa di pietre is a backdrop in the aperta savana.
The alberi di acacia e cespugli spinosi is a backdrop in the aperta savana. "Cespugli spinosi e acacie, non molto fitti, ricoprono la savana in ogni direzione."
The villaggio di capanne is a backdrop in the aperta savana. "Poche capanne in legno ricoperte di frasche secche."

The leone is an animal in the aperta savana. "C[']è un [o]leone[x] sdraiato sulla via per il [s]villaggio[x]. Deve aver appena mangiato e non sembra pericoloso, ma sbarra la strada per il

villaggio." The description is "[mostra-leone-sdraiato]Un leone maestoso, con una criniera ben curata. Non fa caso a te perché è sazio e le scimmie nude non gli sono mai piaciute. E['] comodo dove si trova, ma basterebbe una formica di troppo per infastidirlo."

To say mostra-leone-in-piedi:
 [%\$] show-side figure of leone-in-piedi; [%\$]
 [%\$] play sound of leone; [%\$]
 do nothing.

To say mostra-leone-sdraiato:
 [%\$] show-side figure of leone-sdraiato; [%\$]
 [%\$] play sound of leone; [%\$]
 do nothing.

The carogna di gnu is a female container in the aperta savana. "Vedi anche la carogna di uno [o]gnu[x] ormai ridotta all[']osso. Deve essere l[']ultimo pasto del leone." The description is "Non c[']è quasi rimasto niente[if the pezzo di carne is in the carogna], se non un ultimo pezzo di [o]carne[x],[end if] su questa carogna di gnu, ma un[']esercito di formiche continua comunque il suo lavoro incessante[colonna-formiche]."

Instead of taking the carogna di gnu:
 say "Fa troppo ribrezzo per poterla sollevare."

The pezzo di carne is a edible thing in the carogna di gnu. The keyword is "carne". The description is "Un bel [o]pezzo[x] di carne tra le ossa spolpate: troppo poco perché il leone se ne interessi, ma una festa per le formiche."

Instead of smelling the carogna di gnu:
 say "Puzza proprio come una... carogna!"

Instead of smelling the pezzo di carne:
 say "Puzza di carogna!"

Instead of tasting the pezzo di carne:
 say "Se non fosse per i vermi, non saprebbe di niente!"

Instead of eating the pezzo di carne for the first time:
 say "Ne mangi un piccolo pezzo e pensi: 'Non male, manca solo un po['] di sale.'"

Instead of eating the pezzo di carne for more than one time:
 say "Ti trattiene perché pensi: 'E['] meglio smettere o non ne rimarrà niente.'"

Instead of eating the striscioline di carne for the first time:
 say "Ti trattiene perché non ce n[']è quasi più."

The striscioline di carne is a female, plural-named edible thing. The description is "Carne di gnu ridotta a brandelli: troppo poco perché il leone se ne interessi, ma una festa per le formiche."

Before cutting the pezzo di carne when the player does not carry the pezzo di carne:
 say "Come puoi tagliare il pezzo di carne se non lo tieni in mano." instead.

Before taking something (called the taken) when the player is carried by a rideable animal (called the mounted):
 say "Sei seduto [su-prep the mounted] e reggi le redini. [s]Scendi[x] prima di prendere [the taken]." instead.

Instead of cutting the pezzo di carne:
 award 10 points;
 remove the pezzo di carne from play;
 now the player carries the striscioline di carne;
 say "[mostra-forbici]Hai tagliato il pezzo di carne in tante [o]striscioline[x]. Ora non ti resta che lanciarle... a cosa?"

Instead of throwing the pezzo di carne at the leone:
 say "[mostra-leone-sdraiato]Ti trattiene perché pensi: 'E['] un lancio inutile: il leone lo ignorerebbe, ma se i pezzetti fossero molti di più le formiche...'"

Instead of throwing the striscioline di carne at the leone:
 now the striscioline di carne are in the aperta savana;
 say "[mostra-leone-sdraiato]Il lancio è ben riuscito ed i pezzetti di carne si sono distribuiti tra la carogna e il leone. Il leone annusa i pezzetti, ma rimane pigramente sdraiato al suo posto. Ti guarda come se si chiedesse: 'Che cos[']è questa elemosina?'"

After doing anything when the striscioline di carne are in the aperta savana:
 if the striscioline di carne has been in the aperta savana for more than three turns:
 award 50 points;
 remove the striscioline from play;
 remove the leone from play;
 say "[mostra-leone-in-piedi]Trascorso il tempo necessario, le formiche, dopo aver

seguito la pista di pezzetti di carne, sono giunte in prossimità del leone. Appena il leone si è accorto della loro fastidiosa presenza, si è alzato pigramente e si è allontanato alla ricerca di una comoda acacia su cui digerire. Ormai lo hai perso di vista e la via è libera.";

continue the action.

The colonna di formiche is a female thing. "Vedi una lunga [s]colonna[x] di formiche che dalla carogna di [o]gnu[x] [if the striscioline di carne are on-stage]si dirige verso il [o]leone[x] [otherwise]fa la spola verso il [s]formicaio[x][end if]." The description is "Migliaia di formiche [if the striscioline di carne are in the aperta savana][mostra-leone-sdraiato]che si avviano inarrestabili verso il leone[otherwise][mostra-formicaio]affaccendate per trasportare minuscoli pezzetti di cibo nel formicaio[end if]."

Instead of taking the colonna di formiche:
say "Prendere le formiche una per una? Che perdita di tempo!"

To say mostra-formicaio:
[%] show-side figure of formicaio; [%]
do nothing.

The formicaio is scenery in the aperta savana. "[mostra-formicaio]Un grosso formicaio che si erge con evidenza dal terreno. Le formiche, nel loro piccolo, trovano il modo per farsi notare."

To say colonna-formiche:
now the colonna di formiche is in the aperta savana.

Instead of going to the villaggio degli agricoltori when the leone is on-stage:
say "[mostra-leone-sdraiato]Purtroppo il leone si trova sull[']unica strada per il villaggio. Anche se è sazio e mezzo addormentato non è opportuno stuzzicarlo."

Chapter 8 - Il villaggio degli agricoltori

The villaggio degli agricoltori is a room. It is south of the aperta savana. "Il villaggio degli agricoltori è costituito da poche [s]capanne[x]. A [d]nord[x] si torna nella [s]savana[x], ad [d]est[x] ci sono i [s]vigneti[x], a [d]sud[x] invece si trovano i [s]campi[x] di grano ed, infine, a [d]ovest[x] c[']è il bosco di [s]acacie[x] con le arnie degli apicoltori." [%] The top-illustration is figure of villaggio-degli-agricoltori. [%]

The cielo is a backdrop in the villaggio degli agricoltori.
The pianura della savana is a backdrop in the villaggio degli agricoltori.
The villaggio di capanne is a backdrop in the villaggio degli agricoltori.
The grappoli di uva nei vigneti sulle colline is a backdrop in the villaggio degli agricoltori. The description is "[mostra-uva]Vigne ben curate, distese sulle colline e cariche di grappoli d[']uva."
The grano in spighe nei campi is a backdrop in the villaggio degli agricoltori. The description is "[mostra-grano]Spighe di grano ricolme di chicchi riempiono la pianura."
The arnie nel bosco delle acacie is a female, plural-named backdrop in the villaggio degli agricoltori. The description is "Arnie ricolme di miele distribuite intorno al bosco di acacie."

To say mostra-uva:
[%] show-side figure of uva; [%]
do nothing.

Chapter 9 - I vigneti

The vigneti delle colline is a plural-named room. It is east of the villaggio degli agricoltori. "I vigneti si estendono a perdita d[']occhio sulle [s]colline[x]. Ovunque puoi vedere grappoli d[']uva[x] maturi ed il profumo del mosto è molto forte.[line break]A [d]ovest[x] torni al [s]villaggio[x] degli agricoltori." [%] The top-illustration is figure of vigneti-delle-colline. [%]

The cielo is a backdrop in the vigneti delle colline.
The grappoli di uva nei vigneti sulle colline is a backdrop in the vigneti delle colline.
The villaggio di capanne is a backdrop in the vigneti delle colline.

The vignaiolo is a man in the vigneti delle colline. "Vedi un [o]vignaiolo[x] che [if the vignaiolo is the current interlocutor]parla con te con le gambe immerse nel mosto fino al ginocchio[otherwise]scalzo pigia l[']uva in un tino[end if]." The description is "[mostra-uva]Un esperto vignaiolo, scalzo, [if the vignaiolo is the current interlocutor]con i piedi nel mosto[otherwise]che pigia l[']uva. Se vuoi parlargli, prova prima a salutarlo con un [t]ciao[x][end if]."

The tino is a open enterable container in the vigneti delle colline. "C[']è un [o]tino[x] pieno di [if pigiato]mosto[otherwise]grappoli d[']uva pronti per essere pigiati[end if]." The description is "[mostra-tino]Il tino è pieno di [if pigiato]mosto[otherwise]grappoli d[']uva[end if]."

Instead of taking the tino:
say "[mostra-tino]E['] troppo pesante. Non si può sollevare."

Instead of inserting something (called the object) into the tino:
say "[mostra-tino]Non metti [the object] nel tino anche se pensi tra te e te: 'Infilarci

[The object] fa sicuramente meno danni che pestarci dentro con un paio di piedi sudati..."

Instead of smelling the tino:

```
say "[mostra-tino][if the tino is pigiato]Il profumo del mosto è inebriante.
[otherwise]L[']odore dell[']uva è invitante.[end if]"
```

The tino can be pigiato. The tino is not pigiato.
The tino can be accessibile. The tino is not accessibile.

The lavoro VGN is discussable. The printed name is "[t]lavoro[x]".
The mosto VGN is discussable. The printed name is "[t]mosto[x]".
The vino VGN is discussable. The printed name is "[t]vino[x]".
The additivo VGN is discussable. The printed name is "[t]additivo[x]".

The lavoro VGN, mosto VGN, vino VGN, additivo VGN are part of the vignaiolo.

vignaiolo-node is a closed convnode.
The node of the vignaiolo is vignaiolo-node.
The ask-suggestions are {lavoro VGN, vino VGN}.
The tell-suggestions are {mosto VGN}.

After saying hello to the vignaiolo when the greeting type is explicit:

```
say "[mostra-uva]'Salute a te!' esordisci.[line break]Il vignaiolo interrompe il suo [t]lavoro[x]
[add lavoro VGN ask suggestion] e risponde: 'Salute!'"
```

After saying hello to the vignaiolo when the greeting type is implicit:

```
say "[mostra-uva]Attiri l[']attenzione con un colpo di tosse. Il vignaiolo ti nota e smette di
pigiare l[']uva nel tino."
```

After saying goodbye to the vignaiolo when the farewell type is explicit:

```
say "'Addio!' concludi.[line break]'Addio!' risponde il vignaiolo e torna a pigiare l[']uva."
```

After saying goodbye to the vignaiolo when the farewell type is implicit:

```
say "Il vignaiolo ti saluta e ricomincia a pigiare l[']uva."
```

Default Response of vignaiolo-node:

```
say "'Hai qualcosa di cui parlare o da chiedere?' dice il vignaiolo con impazienza.[line break]";
try silently listing suggested topics.
```

Response of vignaiolo-node when asked about the lavoro VGN:

```
say "'Produci vino?' chiedi pleonasticamente.[paragraph break]'No,' risponde sarcasticamente, 'mi
piace lavarmi i piedi con il vino.'[remove lavoro VGN ask suggestion]"
```

Response of vignaiolo-node when asked about the vino VGN:

```
now the tino is accessibile;
say "[mostra-tino]'Hai una botticella di vino da darmi?' chiedi mostrando le mani per far capire che
non puoi pagare.[paragraph break]'Certamente!' dice il vignaiolo, 'Se mi pigi l[']uva di quel
[o]tino[x], ti darò come ricompensa una botticella di vino novello.'[remove vino VGN ask
suggestion]"
```

Response of vignaiolo-node when told about mosto VGN:

```
say "[mostra-tino]'Il mosto dei tini è molto profumato e frizzante,' affermi.[paragraph break]Il
vignaiolo risponde: 'E['] tutto merito dell['][t]additivo[x] segreto.'[remove mosto VGN tell
suggestion][add additivo VGN ask suggestion]"
```

Response of vignaiolo-node when asked about the additivo VGN:

```
say "'Un additivo segreto? Di che si tratta?' chiedi con ingenua curiosità.[paragraph break]'Te ne
potrei parlare... se non fosse segreto. L[']unica cosa che ti posso dire e che ha a che fare con i
piedi, ma non chiedermi altro,' risponde il vignaiolo.[paragraph break]Dopo quella rivelazione ti è
comunque passata la voglia di fare altre domande sulla produzione del vino.[remove additivo VGN ask
suggestion]"
```

Instead of entering the tino when the tino is not accessibile:

```
say "[mostra-tino]Il [o]vignaiolo[x] ti grida: 'Ehi! Non vorrai entrare nel tino senza il
mio permesso?!'";
```

To say mostra-tino:

```
[$%] show-side figure of tino; [%$]
do nothing.
```

Instead of entering the tino when the tino is accessibile and the player is carried by a rideable animal (called the mounted):

```
say "[mostra-tino]Il vignaiolo ti lancia un urlo: 'Ehi! Non vorrai entrare nel tino in
groppe [a-prep the mounted]! Scendi.'";
```

Instead of entering the tino when the tino is accessibile and the player wears the sandali:

```
say "[mostra-tino]Il vignaiolo ti apostrofa: 'Ehi! Non vorrai entrare nel tino con i
[o]sandali[x]?! Togliteli.'";
```

```

Instead of entering the tino when the player does not wear the sandali:
  if the tino is not pigiato:
    award 5 points;
    now the tino is pigiato;
    now the player carries the botticella di vino;
    say "[mostra-tino]Ti eri sempre chiesto che effetto faceva pigiare l[']uva ed ora lo
sai. Dopo due ore buone di pigia pigia, l[']uva è completamente trasformata in mosto ed il
vignaiolo, soddisfatto, ti regala una [o]botticella[x] di vino novello.[paragraph break]'Torna
quando vuoi. C[']è sempre bisogno di piedi buoni,' dice con involontaria ironia.";
  otherwise:
    say "[mostra-tino]E['] inutile entrarci. I grappoli nel tino sono già stati pigiati."

```

The botticella di vino is a female thing. The description is "Una piccola botte di vino novello. Profuma di mosto e si sente frizzare all[']interno." The printed name is "[o]botticella[x] di vino".

```

Instead of smelling the botticella di vino:
  say "Profuma di mosto!"

```

```

Instead of tasting the botticella di vino:
  say "Ha il buon gusto del vino novello!"

```

```

Instead of listening to the botticella di vino:
  say "Senti il vino frizzare!"

```

```

Instead of drinking the botticella di vino for the first time:
  say "Ne bevi un piccolo sorso e il vino ti dà subito alla testa."

```

```

Instead of drinking the botticella di vino for more than one time:
  say "Ti trattiene perché pensi: 'E['] meglio smettere o non ne rimarrà un goccio.'"

```

Understand "botte" as botticella di vino.

Chapter 10 - I campi di grano

The campi di grano is a plural-named room. It is south of the villaggio degli agricoltori. "I campi di [s]grano[x] si estendono a perdita d[']occhio nella pianura. Ovunque puoi vedere [s]spighe[x] mature ed il profumo del fieno è molto forte.[line break]A [d]nord[x] torni al [s]villaggio[x] degli agricoltori." [%] The top-illustration is figure of campi-di-grano. [%]

The cielo is a backdrop in the campi di grano.
 The grano in spighe nei campi is a backdrop in the campi di grano.
 The villaggio di capanne is a backdrop in the campi di grano.

The contadino is a man in the campi di grano. "Vedi un [o]contadino[x] che [if the contadino is the current interlocutor]parla con te appoggiato al[otherwise]miette il grano con [end if]la sua falce. Ha un aspetto diverso rispetto agli altri abitanti della regione, infatti è biondo e di pelle molto chiara." The description is "[mostra-grano]Un esperto contadino, munito di falce[if the contadino is the current interlocutor] che ti sta parlando[otherwise]. Se vuoi parlargli, prova prima a salutarlo con un [t]ciao[x][end if]."

```

To say mostra-grano:
  [%] show-side figure of grano; [%]
  do nothing.

```

The lavoro CNT is discussable. The printed name is "[t]lavoro[x]".
 The falce CNT is a female discussable. The printed name is "[t]falce[x]".
 The provenienza CNT is a female discussable. The printed name is "[t]provenienza[x]".
 The Islanda CNT is a female discussable. The printed name is "[t]Islanda[x]".

The lavoro CNT, falce CNT, provenienza CNT, Islanda CNT is part of the contadino.

contadino-node is a closed convnode.
 The node of the contadino is contadino-node.
 The ask-suggestions are {lavoro CNT, provenienza CNT}.
 The tell-suggestions are {falce CNT}.

```

After saying hello to the contadino when the greeting type is explicit:
say "[mostra-grano]'Salute a te!' esordisci.[line break]Il contadino interrompe il suo [t]lavoro[x]
[add lavoro CNT ask suggestion] e risponde: 'Salute!'"

```

```

After saying hello to the contadino when the greeting type is implicit:
say "[mostra-grano]Attiri l[']attenzione con un colpo di tosse. Il contadino ti nota e smette di
falciare."

```

```

After saying goodbye to the contadino when the farewell type is explicit:
say "'Addio!' concludi.[line break]'Addio!' risponde il contadino e torna a falciare."

```

```

After saying goodbye to the contadino when the farewell type is implicit:
say "Dopo averti salutato, il contadino riprende a falciare."

```

Default Response of contadino-node:

```
say "'Hai qualcosa di cui parlare o da chiedere?' dice il contadino con impazienza.[line break]";
try silently listing suggested topics.
```

Response of contadino-node when asked about the lavoro CNT:

```
say "'Di cosa ti occupi?' chiedi per attaccare discorso.[paragraph break]'Non lo vedi? Falcio il grano,' ti risponde scioccato.[remove lavoro CNT ask suggestion]".
```

Response of contadino-node when asked about the provenienza CNT:

```
say "'Da dove vieni? Sei molto diverso dagli altri abitanti di questa regione.'[paragraph break]Il contadino diventa triste e ti racconta: 'Ho i capelli biondi e la pelle chiara perché sono di origine nordica. Tanti anni fa ho lasciato la mia isola, l['][t]Islanda[x], perché il clima del mio paese era troppo freddo per me, ma ho ancora una grande nostalgia della mia terra.'[remove provenienza CNT ask suggestion][add Islanda CNT ask suggestion]".
```

Response of contadino-node when asked about the Islanda:

```
say "'Vuoi dire che vieni proprio dall[']Islanda e parli islandese?' chiedi con stupore.[paragraph break]Il contadino ti risponde: 'Sì, vengo dall[']Islanda e parlo islandese da quando sono nato, ma non mi è molto utile da queste parti. Non c[']è molta gente che chiede traduzioni dall[']islandese.'[remove Islanda CNT ask suggestion]".
```

Response of contadino-node when told about falce:

```
say "'Sei davvero bravo nel maneggiare la falce,' affermi con ammirazione.[paragraph break]'E['] una dote naturale,' ti risponde.[remove falce CNT tell suggestion]".
```

Instead of giving the libretto di istruzioni to the contadino:

```
say mostra-grano;
if the libretto is not tradotto;
now the libretto is tradotto;
award 25 points;
say "'Uh!? Un libretto scritto in islandese. E['] tanto tempo che non leggo qualcosa nella mia lingua.'[paragraph break]Sfoggia alcune pagine con attenzione, poi ti spiega: 'Si tratta di una serie di istruzioni per far funzionare un tappeto volante. Dice:[paragraph break][italic type]Tappeto volante per asini (solo un asino infatti potrebbe credere che un tappeto può volare)[line break]Istruzioni: mettete l[']asino sopra il tappeto e un po['] di terra del luogo di destinazione sotto il tappeto. Subito l[']asino volerà verso la meta prescelta.[roman type][paragraph break]'Non sembra molto complicato,' commenta il contadino restituendoti il libretto.";
otherwise:
say "L[']islandese prende il libretto ma te lo restituisce subito dicendo: 'L[']ho già letto. Dice:[paragraph break][italic type]Tappeto volante per asini (solo un asino infatti potrebbe credere che un tappeto può volare)[line break]Istruzioni: mettete l[']asino sopra il tappeto e un po['] di terra del luogo di destinazione sotto il tappeto. Subito l[']asino volerà verso la meta prescelta.[roman type]'[line break]".
```

Understand "islandese" as contadino.

Chapter 11 - Il bosco di acacie

After going to the bosco di acacie:

```
now the bomboletta spray is not agitato;
continue the action.
```

The bosco di acacie is a room. It is west of the villaggio degli agricoltori. "Il bosco di acacie è circondato dalle [s]arnie[x] degli apicoltori. Ovunque puoi vedere favi attorniate dall[']operosità delle [s]api[x] ed il profumo del miele è molto forte.[line break]Ad [d]est[x] torni nel [s]villaggio[x] degli agricoltori." [%\$] The top-illustration is figure of bosco-di-acacie. [%\$]

The cielo is a backdrop in the bosco di acacie.

The api feroci is a female, plural-named backdrop in the bosco di acacie. The description is "Numerose api volano in tutte le direzioni."

The arnie nel bosco delle acacie is a backdrop in the bosco di acacie.

The villaggio di capanne is a backdrop in the bosco di acacie.

The arnia di apis mellifica ferox is a female, closed, openable container. The keyword is "arnia". The description is "[mostra-arnia]Un[']arnia [if open]aperta[otherwise]chiusa[end if] di [italic type]apis mellifica ferox[roman type]." The printed name is "[o]arnia[x] di [italic type]apis mellifica ferox[roman type]".

To say mostra-arnia:

```
[%$] show-side figure of arnia; [%$]
do nothing.
```

Instead of taking the arnia di apis mellifica ferox:

```
say "[mostra-arnia]L[']arnia non si può sollevare, è troppo pesante."
```

Instead of listening to the arnia di apis mellifica ferox:

```
say "[mostra-arnia]Più che un tenue ronzio senti un preoccupante rombo di api inferocite."
```

After opening the arnia di apis mellifica ferox:

```
say "[mostra-arnia]Uno [o]sciame[x] di api inferocite esce dall[']arnia [sciame-ferox]e si mette alla caccia del disturbatore della quiete dell[']alverare. Fuggi e, mantenendoti a debita distanza, osservi con attenzione il volo irregolare delle api."
```

Instead of closing the arnia di apis mellifica ferox when the sciame di api inferocite is in the location:

```
say "[mostra-arnia]Non riesci ad avvicinarti all[']arnia perché lo sciame ti assale minaccioso."
```

Instead of taking the favo when the sciame di api inferocite is in the location:

```
say "[mostra-arnia]Non riesci ad avvicinarti all[']arnia perché lo sciame ti assale minaccioso."
```

After spraying the bomboletta spray on the sciame di api inferocite:

```
award 50 points;
remove sciame di api inferocite from play;
say "[mostra-spray]Le api vengono investite dallo spray e subito il repellente ha effetto. Lo sciame si disperde velocemente e, dopo pochi istanti, non ce n[']è più traccia. Ha funzionato! [paragraph break]L[']apicoltore stupefatto si complimenta con te per l[']impresa portata a termine e si dimostra interessato alla bomboletta di repellente. Dopo una breve contrattazione vi accordate per uno scambio: se gli cedi la [o]bomboletta[x], l[']apicoltore ti permette di prendere il [o]favo[x] di miele dell[']arnia che hai liberato."
```

Instead of taking the favo when the sciame di api inferocite is off-stage and the bomboletta spray is on-stage:

```
say "[mostra-favo]Prima di lasciarti prendere il [o]favo[x], l[']apicoltore vuole la [o]bomboletta[x]."
```

Instead of giving the bomboletta spray to the apicoltore:

```
award 10 points;
remove the bomboletta spray from play;
say "[mostra-spray]L[']apicoltore ringrazia, si toglie il velo anti-punture e comincia a spruzzarsi il repellente addosso."
```

The sciame di api inferocite is an animal. "Uno [o]sciame[x] di api inferocite ronza minaccioso intorno all[']arnia." The description is "[mostra-favo]Chi capita a tiro di queste api inferocite farà una brutta fine."

To say sciame-ferox:

```
now the sciame di api inferocite is in the bosco di acacie.
```

The favo is a thing in the arnia di apis mellifica ferox. The description is "[mostra-favo]Un favo ricolmo di miele e ricoperto di api operose."

The apicoltore is a man in the bosco di acacie. "Vedi un [o]apicoltore[x] coperto da un sottile velo anti-punture [if the apicoltore is the current interlocutor]che parla con te[otherwise]intento a lavorare su un[']arnia[end if]." The description is "[mostra-arnia]Un esperto allevatore d[']api, ricoperto da un velo anti-punture [if the apicoltore is the current interlocutor]che parla con te[otherwise]intento a lavorare su un[']arnia. Se vuoi parlargli, prova prima a salutarlo con un [t]ciao[x][end if]."

The miele APC is discussable. The printed name is "[o]miele[x]".

The lavoro APC is discussable. The printed name is "[t]lavoro[x]".

The aiutante APC is a discussable. The printed name is "[t]aiutante[x]".

The protezione APC is a female discussable. The printed name is "[t]protezione[x]".

The miele APC, lavoro APC, aiutante APC, protezione APC is part of the apicoltore.

apicoltore-node is a closed convnode.

The node of the apicoltore is apicoltore-node.

The ask-suggestions are {lavoro APC}.

The tell-suggestions are {miele APC}.

After saying hello to the apicoltore when the greeting type is explicit:

```
say "[mostra-arnia]'Salute a te!' esordisci.[line break]L[']apicoltore interrompe il suo [t]lavoro[x][add lavoro APC ask suggestion] e risponde: 'Salute!'"
```

After saying hello to the apicoltore when the greeting type is implicit:

```
say "[mostra-arnia]Attiri l[']attenzione con un colpo di tosse. L[']apicoltore ti nota ed interrompe il suo lavoro."
```

After saying goodbye to the apicoltore when the farewell type is explicit:

```
say "'Addio!' concludi.[line break]'Addio!' risponde l[']apicoltore e torna al suo lavoro."
```

After saying goodbye to the apicoltore when the farewell type is implicit:

```
say "L[']apicoltore ti saluta e riprende il suo lavoro."
```

Default Response of apicoltore-node:

```
say "'Hai qualcosa di cui parlare o da chiedere?' dice l[']apicoltore con impazienza.[line break]";
try silently listing suggested topics.
```

Response of apicoltore-node when asked about the lavoro APC:

```
say "'E['] un lavoro molto pericoloso?' chiedi con un po['] di apprensione.[paragraph break]'No,'
risponde, 'basta mantenere la calma e avere... una [t]protezione[x] adeguata.'[remove lavoro APC ask
suggestion][add protezione APC ask suggestion]"
```

Response of apicoltore-node when asked about the protezione APC:

```
say "'Che tipo di protezione si può considerare adeguata?' chiedi preoccupato.[paragraph break]L[']
apicoltore risponde: 'Diciamo che [if the player is nudo]così come sei combinato, la protezione è
insufficiente.'[otherwise]il grembiule che indossi potrebbe essere abbastanza spesso... se non fosse
aperto dietro.'[end if][remove protezione APC ask suggestion]"
```

Response of apicoltore-node when told about the aiutante APC:

```
say "'Il posto da aiutante mi interessa!' affermi con entusiasmo.[paragraph break]Sul volto
dell[']apicoltore si disegna un mezzo sorriso beffardo e dice: 'Molto bene! Per ottenere il posto è
sufficiente che superi una prova molto semplice. In quell[']arnia laggiù...', dice indicando
un[']arnia isolata dalle altre[arnia-ferox], 'c[']è una colonia di [italic type]apis mellifica
ferox[roman type]. Apri l[']arnia, estrai un favo e me lo porti. Se riesci in questa semplice
operazione, avrai il posto.'[paragraph break]Chiedi: '[italic type]Apis mellifica[roman
type]...?'[paragraph break]'... [italic type]ferox[roman type],' completa l[']apicoltore. Dopo
queste parole, ti avvii titubante verso l['][o]arnia[x] mentre l[']apicoltore si siede per assistere
comodamente alla scena.[remove aiutante tell suggestion]"
```

To say arnia-ferox:

```
now the arnia di apis mellifica ferox is in the bosco di acacie.
```

Response of apicoltore-node when told about miele APC:

```
say "[mostra-favo]'Sembra che qui si produca molto miele,' dici all[']apicoltore.[paragraph
break]'Troppo,' risponde, 'tanto che avrei bisogno di un [t]aiutante[x], ma non è facile trovare
operai competenti in materia.'[remove miele APC tell suggestion][add aiutante APC tell suggestion]"
```

```
[ FINE ]
```